Evaluating the Alignment of the Polarized Radio Waves from 27 QSOs in a Region near the NGP

Richard Shurtleff *

Abstract

The sample of 27 quasars with polarized radio emissions located in a region near the North Galactic Pole is shown to have highly aligned polarization directions. Furthermore, by extending their polarization directions around the Celestial Sphere, the convergence of their polarization directions is shown to be close to the sources. Thus, parallax forces the position angles to vary with locations of individual sources. One suspects that, whatever physical explanation fits, the explanation for converging close to the sample is different from the explanation for alignments with near-equal position angles that converge far from the sample on the sky. The alignment is analyzed in this Mathematica notebook. Access to a .nb notebook is provided in the references.

Keywords: Polarized Radio Sources; Alignment; Quasi-stellar objects

*Department of Sciences, Wentworth Institute of Technology, 550 Huntington Avenue, Boston, MA, USA, 02115, orcid.org/0000-0001-5920-759X, e-mail addresses: shurtleffr@wit.edu, momentummatrix@yahoo.com

```
In[ ]:= Print["The date and time that this statement was evaluated: ", Now]

The date and time that this statement was evaluated:  Mon 10 May 2021 06:03:02 GMT−4.
```

0. Preface

The pdf version of this notebook is available online from the viXra archive.
To find the ready-to-run notebook follow the link in Ref. 1.

Notes:

(1) The pdf version quotes some numerical values that are associated with the particular settings and uncertainty runs that were current when the pdf version was created. Other sets of uncertainty runs, for a sufficiently large number of runs, should alter those numerical values only slightly.

(2) The notebooks in this series were created using Wolfram Mathematica, Version Number: 12.1, Ref. 2.

(3) The formulas for creating Aitoff plots were found on Wikipedia, Ref. 3.

The Hub Test

This notebook presents an application of the Hub Test, which is discussed more fully in Ref. 4. The basic idea is that polarization directions are well-aligned with each other when they are well-aligned with some point on the Celestial Sphere.

Consider the well-known prescription for finding Polaris, the North Star, based on the alignment of the direction from the Merak to Dubhe with Polaris. Guided by Fig. 1, let the source S be the star Merak, take the interval from Merak to Dubhe in place of the direction of polarization $\hat{v}_{\psi}$, and let Polaris be the point H. Then the alignment of the Merak to Dubhe direction $\hat{v}_{\psi}$ with Polaris, the point H, illustrates the concept of alignment in the Hub Test. With Merak as S, Merak-Dubhe as $\hat{v}_\psi$, and Polaris as H, the angle η would be about $\eta = 3.47°$. In that case, the blue great circle and the purple great circle in Fig. 1 would almost coincide.

Out[ ]=



Figure 1: The Celestial sphere is pictured on the left and on the right is the plane tangent to the sphere at the source S. The linear polarization direction $\hat{v}_\psi$ lies in the tangent plane and determines the purple great circle on the sphere. A point H on the sphere and the point S determine a second great circle, the blue circle drawn on the sphere at the left. Clearly, H and S must be distinct in order to determine a great circle.

In Fig. 1, the "alignment angle" η is the acute angle η between the great circles at S, $0° \leq \eta \leq 90°$. The alignment angle η measures how well the polarization direction $\hat{v}_\psi$ matches the direction toward the point H. Perfect alignment occurs when $\eta = 0°$ and the two great circles overlap. Perpendicular great circles, $\eta = 90°$, indicates maximum "avoidance" of the polarization direction $\hat{v}_\psi$ with the point H on the sphere. The halfway value, $\eta = 45°$, favors neither alignment nor avoidance.

With N sources S_i, $i = 1, ..., N$, there are N alignment angles η_{iH} for the point H and an average alignment angle $\overline{\eta}$ at H,

$$\overline{\eta}(\text{H}) = \frac{1}{N} \sum_{i=1}^{N} \eta_{\text{iH}} \ . \quad (1)$$

The alignment angle $\overline{\eta}$(H) is a function of position H on the sphere. It is symmetric across diameters, $\overline{\eta}(\text{H}) = \overline{\eta}(\text{-H})$, because great circles are symmetric across diameters.

The function $\overline{\eta}$(H) measures convergence and divergence of the great circles determined by the polarization directions. For random polarization directions, the average $\overline{\eta}$(H) should be near 45°, since each alignment angle η_{iH} is acute, $0° \leq \eta_{\text{iH}} \leq 90°$, and random polarization directions should not favor any one value. Points H where the alignment angle $\overline{\eta}$(H) is smaller than 45°, the great circles tend to converge, where $\overline{\eta}$(H) is larger than 45°, the great circles can be said to diverge.

Thus the basic concept includes "avoidance", as well as alignment. Avoidance is high when the two directions $\hat{v}_\psi$ and $\hat{v}_H$ differ by a large angle, $\eta \to 90°$. Perpendicular great circles at S , $\eta = 90°$, would indicate the maximum avoidance of the polarization direction and the point on the sphere. The N sources' polarization directions most avoid the points H_{max} and $-H_{\text{max}}$ where the function $\overline{\eta}$(H) takes its maximum value $\overline{\eta}_{\text{max}}$. The locations of the most extreme divergence are called "avoidance hubs".

The N sources' polarization directions are best aligned with the points H_{min} and $-H_{\text{min}}$ where the alignment angle is a minimum $\overline{\eta}_{\text{min}}$. The locations H_{min} and $-H_{\text{min}}$ of their most extreme convergence are called "alignment hubs". Alignment and avoidance are equally viable, complementary concepts with the Hub Test.

The Hub test provides many calculated results to describe the collective behavior of the polarization directions in a sample. The alignment angle function $\overline{\eta}$(H), Eq. (1), can be mapped on the Celestial Sphere to give a visual display. The smallest alignment angle $\overline{\eta}_{\text{min}}$ and the largest avoidance angle $\overline{\eta}_{\text{max}}$ quantify the agreement of the directions. Known formulas, see Sec. 4 below, are available to calculate the significance of the alignment, *i.e.* the likelihood that random polarization directions would yield better results. The locations of the convergence hubs H_{min} and the divergence hubs H_{max} may provide clues to magnetic field direction and such quantities.

CONTENTS

In[ ]:=

1. Introduction

Electromagnetic radiation from QSOs has traveled a long way and, no doubt, has been passed along by various intergalactic media. Alternatively, it may be that the radio waves are polarized when emitted at the source. Either way, to have regions of the sky containing QSOs with aligned polarization, or some other way correlated, is certainly remarkable. It has been suggested, Ref. 5, that the polarization levels are too strong, a percent to several percent, for the cause to be local to the Milky Way. With the Coma Supercluster in the same general direction as these QSOs, some mechanism may be able to explain the alignment as occurring enroute, or, as mentioned before, the polarizations could exist when the radio waves are emitted and then some other mechanism would be needed. In any case, the alignment is remarkable.

In this notebook, we analyze the alignment tendencies of the sample of 27 radio QSOs. The sample occupies a roughly 11° radius patch of sky centered on (RA,dec) = (178°,10°) and is chosen based on a whole-sky survey of the radio QSOs in the Pelgrims 2014 catalog, Ref. 6. The survey populated 5°-radius regions centered on the grid points of a 2° mesh and calculated the significance of each region's polarization direction alignment. See Fig. 3. The group that contains the 27 QSOs that are analyzed in this notebook consists of 14 very significantly aligned 5°-radius regions near the North Galactic Pole, one of which happens to be the most significantly aligned of all of the 5° regions.

The 27 QSOs in the sample make 27 great circles along the polarization directions. The smallest alignment angle $\overline{\eta}(H)$ occurs for at a hub H_{min} less than 15° southeast from the center of the sample. When the hub is this close, the polarization directions from different places in the sample must have different position angles due to parallax. The hub test has the advantage that it can detect such correlations.

2. Coordinates, grid, and sundry basic formulas

2a. Coordinates

Consider the "Celestial Sphere", a sphere in 3 dimensional Euclidean space. See Fig. 1 in the Preface. The sphere is also called the "sphere" or sometimes "the sky". The center of the sphere is the origin of a 3D Cartesian coordinate system with coordinates (x, y, z). The direction of the positive z -axis is due "North". Equatorial longitude is the Right Ascension α and latitude is the declination δ.

From a point-of-view located outside the sphere, as in the sketch in Fig. 1, one pictures a source S plotted on the sphere and, in the 2D tangent plane at S, local North is upward and local East is to the right. A "position angle" at the point S on the sphere, such as the angle ψ in Fig. 1, is measured in the 2D plane tangent to the sphere at S. In the tangent plane as drawn in Fig.1, the position angle ψ is measured clockwise from local North with East to the right.

Definitions:

er, eN, eE are unit vectors in a 3D Cartesian coordinate system
(α,δ) = equatorial coordinates longitude and latitude
er(α,δ) = radial unit vectors from Origin
eN(α,δ) = local North at a point on the Celestial Sphere
eE(α,δ) = local East at a point on the Celestial Sphere
αFROMr(er) = α determined by radial unit vector er
δFROMr(er) = δ determined by radial unit vector er

Aitoff Plot Functions
αH(α,δ) , xH(α,δ) , yH(α,δ), where xH is centered on α = 0 and α increases from left-to-right, with α = -180° on the left and +180° on the right
xH180(α,δ) , yH180(α,δ), where xH is centered on α = 180° and α increases from left-to-right, with α = 0° on the left and 360° on the right

In[∘]:=
```
(* For a Source at (α,δ) = (α,δ): er, eN,
eE are unit vectors from Origin to Source, local North, local East, resp.  *)
er[α_, δ_] := er[α, δ] = {Cos[α] Cos[δ], Sin[α] Cos[δ], Sin[δ]}
eN[α_, δ_] := eN[α, δ] = {-Cos[α] Sin[δ], -Sin[α] Sin[δ], Cos[δ]}
eE[α_, δ_] := eE[α, δ] = {-Sin[α], Cos[α], 0}
{"Check er.er = 1, er.eN = 0, er.eE = 0, eN.eN
   = 1, eN.eE = 0,eE.eE = 1, erXeE = eN, eEXeN = er, eNXer = eE: ",
 {0} == Union[Flatten[Simplify[{er[α, δ].er[α, δ] - 1, er[α, δ].eN[α, δ], er[α, δ].eE[α, δ],
      eN[α, δ].eN[α, δ] - 1, eN[α, δ].eE[α, δ], eE[α, δ].eE[α, δ] - 1, Cross[er[α, δ], eE[α, δ]] -
       eN[α, δ], Cross[eE[α, δ], eN[α, δ]] - er[α, δ], Cross[eN[α, δ], er[α, δ]] - eE[α, δ]}]]]}
```

Out[∘]=
```
{Check er.er = 1, er.eN = 0, er.eE = 0, eN.eN = 1,
  eN.eE = 0,eE.eE = 1, erXeE = eN, eEXeN = er, eNXer = eE: , True}
```

Get (α,δ) in radians from a radial vector r:

In[∘]:=
```
αFROMr[r_] := N[ArcTan[Abs[r[[2]] / r[[1]]]]] /; (r[[2]] ≥ 0 && r[[1]] > 0)
αFROMr[r_] := N[π - ArcTan[Abs[r[[2]] / r[[1]]]]] /; (r[[2]] ≥ 0 && r[[1]] < 0)
αFROMr[r_] := N[π + ArcTan[Abs[r[[2]] / r[[1]]]]] /; (r[[2]] < 0 && r[[1]] < 0)
αFROMr[r_] := N[2. π - ArcTan[Abs[r[[2]] / r[[1]]]]] /; (r[[2]] < 0 && r[[1]] > 0)
αFROMr[r_] := π / 2. /; (r[[2]] ≥ 0 && r[[1]] == 0)
αFROMr[r_] := 3 π / 2. /; (r[[2]] < 0 && r[[1]] == 0)
```

In[∘]:=
```
δFROMr[r_] := N[ArcTan[r[[3]] / (√ (r[[1]]^2 + r[[2]]^2))]] /; (√ (r[[1]]^2 + r[[2]]^2) > 0)
δFROMr[r_] := Sign[r[[3]]] (π / 2.) /; (√ (r[[1]]^2 + r[[2]]^2) == 0)
```

The following Aitoff Plot formulas can be found in Wikipedia, Ref. 3.
For these formulas the angles α and δ should be in degrees.
They give an Aitoff Plot that is centered on (0°,0°)

In[∘]:=
```
αH[α_, δ_] := αH[α, δ] = ArcCos[Cos[((2. π) / 360.) δ] Cos[((2. π) / 360.) α / 2.]]
xH[α_, δ_] := xH[α, δ] = (2. Cos[((2. π) / 360.) δ] Sin[((2. π) / 360.) α / 2.]) / Sinc[αH[α, δ]]
yH[α_, δ_] := yH[α, δ] = Sin[((2. π) / 360.) δ] / Sinc[αH[α, δ]]
```

Using the following functions produces an Aitoff Plot that is centered on (180°,0°)

In[∘]:=
```
xH180[α_, δ_] :=
 xH180[α, δ] = (2. Cos[((2. π) / 360.) δ] Sin[((2. π) / 360.) (α - 180.) / 2.]) / Sinc[αH[(α - 180.), δ]]
yH180[α_, δ_] := yH180[α, δ] = Sin[((2. π) / 360.) δ] / Sinc[αH[(α - 180.), δ]]
```

2b. Grid, sometimes called a mesh

We avoid bunching at the poles by taking into account the diminishing radii of constant latitude circles as the latitude approaches the poles. Successive grid points along any latitude or along any longitude make an arc that subtends the same central angle dθ.

We grid one hemisphere at a time, then the grids are combined.

Definitions:

gridSpacing   separation in degrees between grid points on and between constant latitude circles

dθ1	grid spacing in radians
idN, ai, ji	dummy indices, ID #s for grid points, longitude, latitude
αpointH,δpointH	α and δ of the grid points H_j
grid, gridN, gridS	tables data associated with grid points, listings are below
nGrid	number of grid points
αGrid	longitudes at the grid points ($-\pi \le \alpha \le +\pi$)
δGrid	latitudes at the grid points ($-\pi/2 \le \alpha \le \pi/2$)
rGrid	radial unit vectors from origin to grid points, in 3D Cartesian coordinates

Tables: ***grid, gridN* and *gridS***

1. sequential point # 2. α index 3. δ index 4. α (rad) 5. δ (rad) 6. Cartesian coordinates of the grid point

In[∘]:=
```
gridSpacing = 2.(*, in degrees.*);
```

In[∘]:=
```
(*KEEP this cell - DO NOT DELETE*)
(*The Northern Grid "gridN". *)
dθ1 = ((2. π) / 360.) gridSpacing;
(*Convert gridSpacing to radians*)gridN = {};
idN = 1;
For[δj = 0., δj < π / (2. dθ1), δj++, δpointH = δj dθ1;
 For[ ai = 0., ai < Ceiling[ ((2. π) / dθ1) (Cos[δpointH] + 0.01)],
  ai++, αpointH = ai dθ1/(Cos[δpointH] + 0.01);
  AppendTo[gridN, {idN, ai, δj, αpointH, δpointH, er[αpointH, δpointH]}];
  idN = idN + 1
 ]]
```

In[∘]:=
```
(*KEEP this cell - DO NOT DELETE*)
(*The Southern Grid "gridS". *)
dθ1 = ((2. π) / 360.) gridSpacing;(*Convert gridSpacing to radians*)
gridS = {}; idN = 1;
For[δj = 1., δj < π / (2. dθ1), δj++, δpointH = -δj dθ1;
 For[ ai = 0., ai < Ceiling[ ((2. π) / dθ1) (Cos[δpointH] + 0.01)],
  ai++, αpointH = ai dθ1/(Cos[δpointH] + 0.01);
  AppendTo[gridS, {idN, ai, δj, αpointH, δpointH, er[αpointH, δpointH]}];
  idN = idN + 1
 ]]
```

In[∘]:=
```
(*KEEP this cell - DO NOT DELETE*)
grid = {}; j = 1;
For[jN = 1, jN ≤ Length[gridN], jN++, AppendTo[grid, {j, gridN[[jN, 2]],
   gridN[[jN, 3]], αFROMr[gridN[[jN, 6]] ], δFROMr[gridN[[jN, 6]] ], gridN[[jN, 6]]}];
 j = j + 1]
For[jS = 1, jS ≤ Length[gridS], jS++, AppendTo[grid, {j, gridS[[jS, 2]],
   gridS[[jS, 3]], αFROMr[gridS[[jS, 6]] ], δFROMr[gridS[[jS, 6]] ], gridS[[jS, 6]]}];
 j = j + 1]
```

In[∘]:=
```
nGrid = Length[grid];
```

In[ ]:=
```
αGrid = Table[grid[[j, 4]] , {j, nGrid}];
δGrid = Table[grid[[j, 5]] , {j, nGrid}];
rGrid = Table[grid[[j, 6]] , {j, nGrid}];
```

2c. The mean and standard deviation are convenient functions. And we identify directories for getting and putting data.

Definitions

mean   the arithmetic average of a set of numbers, $\frac{1}{N}\sum_{i=1}^{N} n_i$

stanDev   the standard deviation. Given a set of N numbers n_i with mean value m, the standard deviation is $\left(\frac{1}{N}\sum_{i=1}^{N}(n_i - m)^2\right)^{1/2}$, the square root of the average of the squares of the differences of the numbers with the mean. Note that we divide by N to get the average of the deviations squared.

catalogDirectory   directory containing the catalog files
homeDirectory   directory containing the notebook and data files

In[ ]:=
```
mean[data_] := (1 / Length[data]) Sum[data[[i4]], {i4, Length[data]}];
(* arithmetic average *)
stanDev[data_] :=
 ((1 / Length[data]) Sum[(data[[i5]] - mean[data])^2, {i5, Length[data]}])^(1/2)
 (*standard deviation*)
```

In[ ]:=
```
catalogDirectory =
  "C:\\Users\\shurt\\Dropbox\\HOME_DESKTOP-0MRE5OJ\\SendXXX_CJP_CEJPetc\\SendViXra\\
    20200715AlignmentMethod\\20200715AlignmentMMAnotebooks";
 (* location of the catalog data file on my computer*)

homeDirectory =
  "C:\\Users\\shurt\\Dropbox\\HOME_DESKTOP-0MRE5OJ\\SendXXX_CJP_CEJPetc\\SendViXra\\
    20200715AlignmentMethod\\20210505AlignmentMethodv4\\20210515Clump1QSOsNearNGP";
 (*The notebook file and data files for this notebook are put in this directory. *)
```

2d. Section Summary

In[ ]:=
```
Print["The grid points are separated by gridSpacing = ",
 gridSpacing, "° arcs along latitude and longitude."]
Print["The number of grid points is ", nGrid, " ."]
```

```
The grid points are separated by gridSpacing = 2.° arcs along latitude and longitude.

The number of grid points is 10518 .
```

3. Polarization and Position Data

3a. Data

The Pelgrims 2014 catalog incorporates data from the large JVAS/CLASS 8.4 Ghz catalog Jackson 2007, Refs. 6 and 7. The Pelgrims 2014 catalog sources were filtered from Jackson 2007 sources by identification as QSOs, for percent polarization, p > 0.6%, for the largest fractional uncertainty in percent polarization, σp/p < 0.6%, and for uncertainty in the polarization position angle $\sigma_\psi < 16°$. The data is converted to convenient units, angles in radians, and reordered in a notebook The result is the basic data file “data00”.

(The files on my computer: 20200713JVAS1450Todata00a.nb, 20200718data08JVAS1450.dat, JVAS_1450A.dat.txt, 20210418Survey1450QSOs.nb .)

Definitions:

data00	the catalog data, Pelgrims 2014
firstClumpQsosIDinData001450	record numbers in the catalog of the QSOs in the sample
nSrc	number of sources
αSrc	right ascension, longitude (radians)
δSrc	declination, latitude (radians)
ψn	PPA, polarization position angle: clockwise from North with East to the right.
$\sigma\psi$n	uncertainty in PPA
percentPol	percentage of linear polarization
rSrc	unit vector from Origin to Sources on Celestial Sphere
eNSrc	Local North at the ith Source
eESrc	Local East at the ith Source
sourceCenter	unit radial vector to the arithmetic center of the sources
angleSourceToCenter	angle from Source to Center

Input Sources: ***data00*** is the data table saved in the file “20200718data08JVAS1450.dat”, created in the notebook “20200713-JVAS1450Todata00a.nb”.

20200718data08JVAS1450.dat = data table called “data00” below.
Notes: Input must be in the correct units, especially angles in radians. The polarization position angle is measured clockwise from local North with East to the Right.

data00:
1.Object #   2. Ra (rad)   3. Dec (rad)   4. ψ (rad)   5. $\sigma\psi$ (rad)   6. z   7. p (%)   8. σp (%)

Catalog data

```
In[ ]:= SetDirectory[
  "C:\\Users\\shurt\\Dropbox\\HOME_DESKTOP-0MRE5OJ\\SendXXX_CJP_CEJPetc\\SendViXra\\20200715
    AlignmentMethod\\20200715AlignmentMMAnotebooks"]
data00 = Get["20200718data08JVAS1450.dat"];
Length[%]

Out[ ]= C:\Users\shurt\Dropbox\HOME_DESKTOP-0MRE5OJ\SendXXX_CJP_CEJPetc\
  SendViXra\20200715AlignmentMethod\20200715AlignmentMMAnotebooks

Out[ ]= 1450
```

```
In[•]:= rai[i_] := rai[i] = data00[[i, 2]] (*RA of ith source*)
deci[i_] := deci[i] = data00[[i, 3]] (*dec*)
ψi[i_] := ψi[i] = data00[[i, 4]] (*PPA,
polarization position angle: clockwise from North with East to the right. *)
σψi[i_] := σψi[i] = data00[[i, 5]]
zi[i_] := data00[[i, 6]] (*redshift found by Pelgrim's using NED*)
ri[i_] := ri[i] = er[rai[i], deci[i]]
(*unit vector from Origin to ith Source on Celestial Sphere*)
vNi[i_] := vNi[i] = eN[rai[i], deci[i]] (*North*)
vEi[i_] := vEi[i] = eE[rai[i], deci[i]] (*East*)
vψi[i_] := vψi[i] = Cos[ψi[i]] vNi[i] + Sin[ψi[i]] vEi[i] (*unit vector in direction of PPA*)
nSxψi[i_] := nSxψi[i] = Sin[ψi[i]] vNi[i] - Cos[ψi[i]] vEi[i] (* r Cross vψ *)
```

Clump 1 QSO data (from 20210418Survey1450QSOs.nb, a survey with 5°-radius regions)

```
In[•]:= firstClumpQsosIDinData001450 = {659, 660, 663, 667, 674, 680, 682, 690, 695, 696, 698,
   707, 712, 714, 718, 720, 721, 727, 728, 731, 734, 744, 746, 751, 752, 762, 764};

In[•]:= (*right ascension in radians*)
αSrc = 10^-6.
   {2940786, 2950332, 2962501, 2977947, 3000259, 3006888, 3013383, 3037854, 3060196,
    3063615, 3077693, 3108571, 3111962, 3114578, 3131037, 3137987, 3138954, 3154756,
    3156278, 3164771, 3173054, 3207036, 3209928, 3222030, 3222168, 3239225, 3245921};

In[•]:= nSrc = Length[αSrc]

Out[•]= 27

In[•]:= (*declination in radians*)
δSrc = 10^-6. {256694, 148170, 219533, 315742, 103421, 291870, 190246, 258405,
    176105, 275734, 85942, 132052, 161164, 173344, 290596, 52995, 32695, 114811,
    73978, 95356, 212862, 148171, 158862, 193466, 109659, 73672, 119278};

In[•]:= (* position angle in radians*)
ψn = 10^-6. {1788962, 1120501, 2185152, 2724459, 2022837,
    2553417, 2045526, 2857104, 1733112, 2485349, 1877974, 2331760,
    2406809, 2277655, 1937315, 1106539, 1799434, 2961824, 2586578, 2912955,
    1925098, 2600541, 2188643, 2352704, 2827433, 1527163, 2905973};
```

In[◦]:=
```
Histogram[ψn (360./2. π), {12}, PlotLabel → "PPA ψ, number ΔR per bin",
 AxesLabel → {"ψ", "ΔR"}, PlotRange → {{0, 200}, Automatic}]
Print["Figure 2. Distribution of position angles for the
   27 polarization directions in the sample. Note the fairly even
   distribution over sixty degrees or so, ψ = 100° to ψ = 160°."]
```

Out[◦]=



Figure 2. Distribution of position angles for the 27 polarization directions in the sample. Note the fairly even distribution over sixty degrees or so, ψ = 100° to ψ = 160°.

In[◦]:=
```
(*uncertainty in ψ in radians*)
σψn = 10^-6· {4242, 252, 2254, 99, 106 992, 51 458, 112 351, 26 729,
    137 622, 18 357, 10 877, 271 821, 37 352, 134 004, 48 856, 98 592, 277 921,
    7249, 5633, 5724, 66 923, 35 001, 138 200, 114 372, 105 062, 7815, 7653};
```

In[◦]:=
```
(* % polarization*)
percentPol = 10^-6·
   {2 386 846, 4 130 478, 2 023 713, 1 658 885, 1 784 232, 1 979 194, 2 210 679, 6 381 769, 5 954 787,
    2 903 853, 3 866 300, 3 070 517, 1 080 690, 1 854 161, 492 130, 2 652 914, 10 217 777, 3 754 306,
    1 874 058, 3 174 907, 604 797, 653 203, 5 457 402, 615 497, 16 210 481, 901 464, 3 306 869};
```

In[◦]:=
```
(* uncertainty in % polarization*)
σpercentPol = 10^-6· {20 249, 2078, 9121, 328, 381 771, 203 679, 496 710, 341 137,
    1 638 906, 106 607, 84 105, 1 669 146, 80 727, 496 898, 48 084, 523 076, 5 679 057,
    54 428, 21 111, 36 344, 80 945, 45 723, 1 508 313, 140 783, 3 405 959, 14 090, 50 611};
```

In[◦]:=
```
(*Redshift*)
redshift =
  10^-6· {867 400, 486 000, 2 125 700, 1 040 000, 2 217 000, 1 996 700, 1 323 900, 603 700, 1 051 400,
    299 000, 1 343 600, 876 100, 695 900, 895 000, 1 061 200, 1 009 800, 2 440 000, 2 180 900,
    1 226 000, 1 300 000, 890 500, 2 359 000, 2 721 600, 1 404 000, 2 078 200, 966 000, 1 189 000};
```

In[◦]:=
```
rSrc = Table[er[ αSrc[[i]], δSrc[[i]] ],  {i, nSrc}];(*calculated from Input.*)
eNSrc  = Table[eN[ αSrc[[i]], δSrc[[i]] ], {i, nSrc}];(*calculated from Input.*)
eESrc  = Table[eE[  αSrc[[i]], δSrc[[i]]  ], {i, nSrc}];(*calculated from Input.*)
```

In[∘]:=
```
sourceCenter0 = 1/nSrc Sum[rSrc[[i]], {i, nSrc}];
sourceCenter = sourceCenter0/(sourceCenter0.sourceCenter0)^(1/2);
(*unit radial vector to the arithmetic center of the sources.*)
angleSourceToCenter = Table[ArcCos[rSrc[[i]].sourceCenter], {i, nSrc}];
```

3a. Section Summary

We consider Quasi-Stellar Objects, QSOs. The data is found in Pelgrims 2014, Ref. 6, a catalog of 1450 QSOs that have been identified as QSOs in the earlier JVAS/CLASS 8.4Ghz catalog Jackson 2007 that has 12700 records. Ref. 7 Then 5° radius regions are constructed, one on each of the 10518 grid points as in Sec. 2b. The 1450 QSOs were assigned to the regions based on location and we calculated the significance of the alignment of the polarization directions for the sources in each region.

The QSOs selected for this notebook satisfied many requirements: (*i*) have 7 or more sources in order to use the significance formulas in Sec. 4 accurately, (*ii*) have longitude RA 165° $\leq \alpha \leq$ 200°, (*iii*) have latitude dec 0° $\leq \delta \leq$ 30°, (*iv*) whose QSOs are very significantly aligned, $S \leq 10^{-2}$. There are 14 regions satisfying (*i*) - (*iv*) containing a total of 27 sources.

Out[∘]=


Figure 3. Survey of polarized radio QSOs. (Equatorial Coordinates, centered at $(\alpha,\delta) = (180°,0°)$, $\alpha = 360°$ on the right.) The 1450 QSOs were grouped into 5° radius regions centered on grid points. Those regions having at least 7 QSOs are plotted as gray dots at the central grid point. Just 35 regions showed very significant alignment, *i.e.* $S \leq 0.01 = 10^{-2}$, or, equivalently, $-\mathrm{Log}_{10} S \geq 2.0$, and these are plotted as color dots. The indicated clump of 14 regions was selected for the analysis. There are 27 QSOs in the combined area of the 14 regions.

In[∘]:=
```
Print["There are ", nSrc, " sources in the sample."]
Print["Check that the Sample obeys the data cuts:"]
Print[
 "Check that the smallest % polarization p in the sample is 0.5% or more. Smallest: ",
 Sort[percentPol][[1]], "% ."]
Print["Check that the largest fractional uncertainty in % polarization, σp/p,
   is less than 0.6 . Largest: ", Sort[σpercentPol / percentPol][[-1]], " ."]
Print["Check that the largest PPA ψ uncertainty σψ is less than 16°. Largest: ",
 Sort[σψn][[-1]] (360./(2. π)), "° ."]
```

There are 27 sources in the sample.

Check that the Sample obeys the data cuts:

Check that the smallest % polarization p in the sample is 0.5% or more. Smallest: 0.49213% .

Check that the largest fractional uncertainty
in % polarization, σp/p, is less than 0.6 . Largest: 0.555802 .

Check that the largest PPA ψ uncertainty σψ is less than 16°. Largest: 15.9237° .

In[∘]:=
```
ListPlot[Table[{αSrc[[j]], δSrc[[j]]} (360./(2. π)), {j, nSrc}],
 PlotRange → {{0, 360}, {-90, 90}},
 Ticks → {Table[{i, i}, {i, 0, 360, 60}], Table[{j, j}, {j, -90, 90, 30}]},
 PlotLabel → "Sources", AxesLabel → {"α, degrees", "δ, degrees"}, PlotStyle → Green]
Print["Figure 4. The locations of the ", nSrc, " QSOs in the sample. "]
Print[
 "Sample Size: The angular separation of the furthest QSO from the sample center is ",
 Sort[angleSourceToCenter][[-1]] (360./(2. π)), "°."]
```

Out[∘]=


Figure 4. The locations of the 27 QSOs in the sample.

Sample Size: The angular separation of the furthest QSO from the sample center is 11.1277°.

4. Probability Distributions and Significance Formulas

4a. Formulas

The problem of “significance” is to determine the likelihood that random polarizations directions would have better alignment or avoidance than the observed polarization directions. To determine the probability distributions and related formulas, in a previous notebook, we made many runs with random data and fit the results.

For samples with randomly directed polarization vectors, the basic formula, Eq. 1, looks like the sum of random numbers each restricted to the range 0 to π. Such random sums can be related to well-known Random Walk scenarios. That connection helps explain the dependence on $\sqrt{N}$ in the formulas below.

Definitions:

norm	a constant used to normalize the distribution so the integral of probability is 1.
probMIN0, probMAX0	probability distributions for alignment (MIN) and avoidance (MAX), functions of η, η_0, σ
ρciaiMIN,MAX	constants used in the formulas to mean η_0 and uncertainty σ
$\sigma\rho$ciaiMIN,MAX	uncertainty σ in the constants used in the formulas to mean η_0 and uncertainty σ
regionRadiusChoices	radii used in random runs performed elsewhere, not in this notebook
regionChoice	determines the best choice for the current sample
rgnRadius	assumed radius of the region for the purpose of selecting the statistics constants c_i and a_i
iρ	dummy variable used to select region radius
ciMIN,MAX and aiMIN,MAX	parameters for statistics formulas for η_0 and σ
η0MIN, MAX	function to estimate mean η_0
σMIN, MAX	function to estimate uncertainty σ
probMIN, probMAX	probability distributions using estimated values of η_0, σ
signiMIN0, signiMAX0	significance as a function of (η, η_0, σ)
signiMIN, signiMAX	significance of η using estimated values of η_0, σ

In[ ]:=
```
(* y = ((η - η0)/σ); dy = dη/σ *)
(* The normalization factor "norm" is needed for the probability density *)
```
$$\text{norm} = \left(\frac{1}{(2\pi)^{1/2}}\,\text{NIntegrate}\left[\left(1+e^{4\,(y-1)}\right)^{-1} e^{-\frac{y^2}{2}},\ \{y,-\infty,\infty\}\right]\right)^{-1};$$
```
norm;(*Constant needed to make the integral
 of the probability distribution equal to unity.*)
```

In[ ]:=
$$\text{probMIN0}[\eta\_,\ \eta0\_,\ \sigma\_] := \left(\frac{\text{norm}}{\sigma\,(2\pi)^{1/2}}\right)\left(1+e^{4\,\frac{(\eta-\eta0-\sigma)}{\sigma}}\right)^{-1} e^{-\frac{1}{2}\left(\frac{\eta-\eta0}{\sigma}\right)^2}$$
```
signiMIN0[η_, η0_, σ_] := NIntegrate[probMIN0[η1, η0, σ], {η1, -∞, η}]
```

In[ ]:=
$$\text{probMAX0}[\eta\_,\ \eta0\_,\ \sigma\_] := \left(\frac{\text{norm}}{\sigma\,(2\pi)^{1/2}}\right)\left(1+e^{-4\,\frac{(\eta-\eta0+\sigma)}{\sigma}}\right)^{-1} e^{-\frac{1}{2}\left(\frac{\eta-\eta0}{\sigma}\right)^2}$$
```
signiMAX0[η_, η0_, σ_] := NIntegrate[probMAX0[η1, η0, σ], {η1, η, ∞}]
```

The significance signiMIN0[η, η0, σ] is the Integral of probMIN0, i.e. signiMIN0 $= \int_{-\infty}^{\eta} P_{MIN}(\eta)\,d\eta$.

The significance signiMAX0[η, η0, σ] is the Integral of probMAX0, i.e. signiMAX0 $= \int_{\eta}^{\infty} P_{MAX}(\eta)\,d\eta$.

The formulas for mean $\eta_0 = \frac{\pi}{4} \pm \frac{c1}{N^{a1}}$ and half-width $\sigma = \frac{c2}{4\,N^{a2}}$ estimate η_0 and σ by functions of the number N of sources. These formulas depend on the size of the region (radius ρ) by the choice of parameters c_i and a_i, $i = 1,2$. The following values for the parameters c_i and a_i are based on random runs. For each combination of $N = \{8,16,32,64,128,181,256,512\}$ and $\rho = \{0°,5°,12°,24°,48°,90°\}$, there were 2000 random runs completed.

A notation conflict between this notebook and the article, Ref. 4, should be noted. We doubled the exponent "a" so $N^{a/2}$ appears in the article, whereas in the formulas here we see N^a. Thus $a \approx 1/2$ here, but the paper has $a_{\text{Article}} \approx 1$. That explains the "/2" in the following arrays.

In[∘]:= ρciaiMIN =

"ρ"	"c1"	"a1"	"c2"	"a2"
90	0.9423	1.0046/2	1.061	0.954/2
48	0.9505	1.0156/2	1.166	0.9956/2
24	0.9235	1.0069/2	1.127	0.964/2
12	0.8912	1.0054/2	1.238	1.021/2
5	0.8363	1.0088/2	1.076	0.940/2
0	0.5031	1.0153/2	1.522	1.053/2

;

In[∘]:= ρciaiMAX =

"ρ"	"c1"	"a1"	"c2"	"a2"
90	0.9441	1.0055/2	1.000	0.931/2
48	0.9572	1.0165/2	1.090	0.958/2
24	0.927	1.0068/2	1.101	0.964/2
12	0.9049	1.0090/2	1.228	1.018/2
5	0.8424	1.0062/2	1.168	0.992/2
0	0.4982	1.0093/2	1.543	1.060/2

;

In[∘]:= ρΔciaiMIN =

"ρ"	"c1"	"a1"	"c2"	"a2"
90	0.0050	0.0036/2	0.026	0.016/2
48	0.0079	0.0057/2	0.016	0.0095/2
24	0.0024	0.0018/2	0.022	0.013/2
12	0.0034	0.0026/2	0.039	0.021/2
5	0.0035	0.0028/2	0.030	0.019/2
0	0.0059	0.0080/2	0.052	0.024/2

;

In[∘]:= ρΔciaiMAX =

"ρ"	"c1"	"a1"	"c2"	"a2"
90	0.0061	0.0044/2	0.038	0.025/2
48	0.0063	0.0045/2	0.026	0.016/2
24	0.011	0.0079/2	0.019	0.011/2
12	0.0069	0.0052/2	0.039	0.022/2
5	0.0038	0.0031/2	0.022	0.013/2
0	0.0058	0.0080/2	0.057	0.025/2

;

In[∘]:=
```
(*The region radius controls the constants ci and ai for statistics in Sec. 4.*)
regionRadiusChoices = {90, 48, 24, 12, 5, 0};(*Do not change this statement*)
regionChoice = 4; (*This is a setting. The choice 24° is 3rd in the list. *)
rgnRadius = regionRadiusChoices[[regionChoice]];
Print["The region radius ρ is set at ", rgnRadius, "°."]
```

The region radius ρ is set at 12°.

In[∘]:=
```
iρ = regionChoice + 1; (* Parameters c_i, a_i, i = 1,2. *)
Print["These constants are for sources confined to regions with radii ρ = ",
 ρciaiMIN[[iρ, 1]], "°."]
{c1MIN, a1MIN, c2MIN, a2MIN} = Table[ρciaiMIN[[iρ, j]], {j, 2, 5}]
{c1MAX, a1MAX, c2MAX, a2MAX} = Table[ρciaiMAX[[iρ, j]], {j, 2, 5}]
```

These constants are for sources confined to regions with radii ρ = 12°.

Out[∘]= {0.8912, 0.5027, 1.238, 0.5105}

Out[∘]= {0.9049, 0.5045, 1.228, 0.509}

In[∘]:=
```
iρ = regionChoice + 1; (* ± uncertainty for the Parameters c_i and a_i, i = 1,2. *)
Print["These uncertainties are for sources confined to regions with radii ρ = ",
 ρciaiMAX[[iρ, 1]], "°."]
{c1MINplusMinus, a1MINplusMinus, c2MINplusMinus, a2MINplusMinus} =
 Table[ρΔciaiMIN[[iρ, j]], {j, 2, 5}]
{c1MAXplusMinus, a1MAXplusMinus, c2MAXplusMinus, a2MAXplusMinus} =
 Table[ρΔciaiMAX[[iρ, j]], {j, 2, 5}]
```

These uncertainties are for sources confined to regions with radii ρ = 12°.

Out[∘]= {0.0034, 0.0013, 0.039, 0.0105}

Out[∘]= {0.0069, 0.0026, 0.039, 0.011}

In[∘]:=

$$\eta0MIN[nSrc\_, c1\_, a1\_] := \frac{\pi}{4} - \frac{c1}{nSrc^{a1}}$$

$$\sigma MIN[nSrc\_, c2\_, a2\_] := \frac{c2}{4\, nSrc^{a2}}$$

In[∘]:=

$$\eta0MAX[nSrc\_, c1\_, a1\_] := \frac{\pi}{4} + \frac{c1}{nSrc^{a1}}$$

$$\sigma MAX[nSrc\_, c2\_, a2\_] := \frac{c2}{4\, nSrc^{a2}}$$

The following probability distributions and significances make use of the above formulas for mean η_0 and half-width σ. They are functions of the alignment angle η and the number of sources N.

In[∘]:=
```
probMIN[η_, nSrc_] := probMIN0[ η, η0MIN[nSrc, c1MIN, a1MIN], σMIN[nSrc, c2MIN, a2MIN] ]
```

In[∘]:=
```
signiMIN[η_, nSrc_] := signiMIN0[η, η0MIN[nSrc, c1MIN, a1MIN], σMIN[nSrc, c2MIN, a2MIN]]
```

In[∘]:=
```
probMAX[η_, nSrc_] := probMAX0[ η, η0MAX[nSrc, c1MAX, a1MAX], σMAX[nSrc, c2MAX, a2MAX] ]
signiMAX[η_, nSrc_] := signiMAX0[η, η0MAX[nSrc, c1MAX, a1MAX], σMAX[nSrc, c2MAX, a2MAX]]
```

4b. Section Summary

```
In[]:= Print["The angular separation of the furthest source from the region center is ",
   Sort[angleSourceToCenter][[-1]] (360./(2. π)), "°.",
   " We choose the statistics constants a_i and c_i, i = 1,2, for
     sources confined to regions with radii ρ = ", ρciaiMIN[[iρ, 1]], "°."]
  Print["The formulas also depend on the number of sources, nSrc = ", nSrc, "."]
  Print["For this sample, but with random polarization directions,
     the random runs give the smallest alignment angle η̄min, η̄min^Random ψ = ",
   η0MIN[nSrc, c1MIN, a1MIN] (360./(2. π)), "° ± ", σMIN[nSrc, c2MIN, a2MIN] (360./(2. π)),
   "°. (Random ψ)"]
  Print["For this sample, but with random polarization directions,
     the random runs give the largest avoidance angle η̄max, η̄max^Random ψ = ",
   η0MAX[nSrc, c1MAX, a1MAX] (360./(2. π)), "° ± ", σMAX[nSrc, c2MAX, a2MAX] (360./(2. π)),
   "°. (Random ψ)"]
```

The angular separation of the furthest source from the region center is 11.1277°. We choose the statistics constants a_i and c_i, i = 1,2, for sources confined to regions with radii ρ = 12°.

The formulas also depend on the number of sources, nSrc = 27.

For this sample, but with random polarization directions, the random runs give the smallest alignment angle $\bar{\eta}_{\min}$, $\bar{\eta}_{\min}{}^{\text{Random}\,\psi}$ = 35.2602° ± 3.29664°. (Random ψ)

For this sample, but with random polarization directions, the random runs give the largest avoidance angle $\bar{\eta}_{\max}$, $\bar{\eta}_{\max}{}^{\text{Random}\,\psi}$ = 54.8311° ± 3.28622°. (Random ψ)

5. Results using the Best Values ψn of the Polarization Directions

"Best" means we use the ψn that were listed in the catalog. We calculate the alignment function $\bar{\eta}$(H) at the grid points H. Given the alignment function $\bar{\eta}$(H), one can find the smallest alignment angle $\bar{\eta}_{\min}$ and the largest avoidance angle $\bar{\eta}_{\max}$ and determine the significances for the alignment and avoidance of the polarization directions.

In Sec. 6 below, we consider other values of the polarization directions that are near the best values, consistent with uncertainty $\sigma\psi$ in the measured values.

5a. The alignment function $\bar{\eta}$(H).

Definitions:

vψSrc	unit vectors along the polarization directions in the tangent planes of the sources
eN	local unit vectors along local North
eE	local unit vectors along local East
jηBarHj	{j,$\bar{\eta}$(H)}, where j is the index for grid point H_j and $\bar{\eta}$(H) is the average alignment angle at H_j. See Eq. (1) in the Introduction.
sortjηBarHj	{j,$\bar{\eta}$(H)}, sorted, with smallest angles $\bar{\eta}$(H) first.
jηBarMin	{j,$\bar{\eta}$(H)}, the j and $\bar{\eta}$ for the smallest value of $\bar{\eta}$(H), best alignment
ηBarMin	the smallest value of $\bar{\eta}$(H), measures alignment of the polarization directions

jηBarMax	$\{j,\overline{\eta}(\mathrm{H})\}$, the j and $\overline{\eta}$ for the largest value of $\overline{\eta}(\mathrm{H})$, most avoided
ηBarMax	the largest value of $\overline{\eta}(\mathrm{H})$, measures avoidance
nSxψn	unit vector, $S_i \times \psi_i$, cross product of the radial vector to the source with the vector in the direction of the polarization
nSxHnj	unit vector, $S_i \times H_j$, cross product of the radial vector to the source with the radial vector to the grid point H_j
ηnHj	alignment angle between source and grid point H_j, see Fig. 1
ηBarHj	alignment angle $\overline{\eta}(H_j)$ between source and grid point H_j, aveαged over all sources
jηBarHj	$\{j, \overline{\eta}(H_j)\}$, the j and $\overline{\eta}$ for grid point H_j
sigηBarMin	significance of the smallest alignment angle
sigrangeηBarMin	get the range of sigs using the plus/minus values on the parameters c_i, a_i
sigSmallηBarMin	the smallest of the values in sigrangeηBarMin
sigBigηBarMin	the largest of the values in sigrangeηBarMin
sigηBarMax	significance of the largest alignment angle (i.e. avoidance)
sigrangeηBarMax	get the range if sigs using the plus/minus values on the parameters c_i, a_i
sigSmallηBarMax	the smallest of the values in sigrangeηBarMax
sigBigηBarMax	the largest of the values in sigrangeηBarMax
αHminDegrees	α of the point $H_{\min}$ where $\overline{\eta}(\mathrm{H})$ is the smallest
δHminDegrees	δ of the point $H_{\min}$ where $\overline{\eta}(\mathrm{H})$ is the smallest
αHmaxDegrees	α of the point $H_{\max}$ where $\overline{\eta}(\mathrm{H})$ is the largest
δHmaxDegrees	δ of the point $H_{\max}$ where $\overline{\eta}(\mathrm{H})$ is the largest

In[ ]:=

```
(* vψ, eN, eE unit vectors in the tangent plane of each source Si,
pointing along the polarization direction, local North,
and local East, respectively. See Fig. 1.*)
vψSrc = Table[Cos[ ψn[[i]] ] eN[ αSrc[[i]], δSrc[[i]] ] +
    Sin[ ψn[[i]] ] eE[ αSrc[[i]], δSrc[[i]] ], {i, nSrc}];

(* Analysis using Eq (5) in Ref. 4 to get η̄(Hj). First ηiH,
cos(ηiH) = |v̂H.v̂ψi|, and then η̄(Hj), by Eq. (1). *)
jηBarHj =
  Table[{j, (1/nSrc) Sum[ArcCos[ Abs[ rGrid[[j]].vψSrc[[i]] / ((rGrid[[j]] - (rGrid[[j]].
                rSrc[[i]]) rSrc[[i]]).(rGrid[[j]] - (rGrid[[j]].rSrc[[i]])
              rSrc[[i]]))^(1/2) ] - 0.000001 ] , {i, nSrc}]}, {j, nGrid}];
 sortjηBarHj = Sort[jηBarHj, #1[[2]] < #2[[2]] &];
jηBarMin = sortjηBarHj[[1]]; (*  {j,η̄(Hj)} for smallest η̄(Hj) *)
ηBarMin = jηBarMin[[2]];
jηBarMax = sortjηBarHj[[-1]]; (*  {j,η̄(Hj)} for largest η̄(Hj) *)
ηBarMax = jηBarMax[[2]] ;
```

```
In[]:= (*Significance of the smallest alignment angle η̄min .*)
sigηBarMin = signiMIN[ηBarMin, nSrc];
sigrangeηBarMin = Sort[Partition[Flatten[Table[
      {signiMIN0[ηBarMin, η0MIN[nSrc, c1MIN + γ1 c1MINplusMinus, a1MIN + α1 a1MINplusMinus],
        σMIN[nSrc, c2MIN + γ2 c2MINplusMinus, a2MIN + α2 a2MINplusMinus]], γ1, α1, γ2, α2},
      {γ1, -1, 1}, {α1, -1, 1}, {γ2, -1, 1}, {α2, -1, 1}] ], 5 ]   ];
{sigrangeηBarMin[[1]], sigrangeηBarMin[[-1]]};
sigSmallηBarMin = sigrangeηBarMin[[1, 1]];
sigBigηBarMin = sigrangeηBarMin[[-1, 1]];
```

```
In[]:= (*Significance of the largest avoidance angle η̄max .*)
sigηBarMax = signiMAX[ηBarMax, nSrc];
sigrangeηBarMax = Sort[Partition[Flatten[Table[
      {signiMAX0[ηBarMax, η0MAX[nSrc, c1MAX + γ1 c1MAXplusMinus, a1MAX + α1 a1MAXplusMinus],
        σMAX[nSrc, c2MAX + γ2 c2MAXplusMinus, a2MAX + α2 a2MAXplusMinus]], γ1, α1, γ2, α2},
      {γ1, -1, 1}, {α1, -1, 1}, {γ2, -1, 1}, {α2, -1, 1}] ], 5 ]   ];
{sigrangeηBarMax[[1]], sigrangeηBarMax[[-1]]};
sigSmallηBarMax = sigrangeηBarMax[[1, 1]];
sigBigηBarMax = sigrangeηBarMax[[-1, 1]];


(* Equatorial coordinates (α,δ) for the hubs Hmin and Hmax .*)
αHminDegrees = αGrid[[ jηBarMin[[1]] ]] (360 / (2 π)); (*Hmin*)
δHminDegrees = δGrid[[ jηBarMin[[1]] ]] (360 / (2 π));

αHmaxDegrees = αGrid[[ jηBarMax[[1]] ]] (360 / (2 π)); (*Hmax*)
δHmaxDegrees = δGrid[[ jηBarMax[[1]] ]] (360 / (2 π));
```

```
In[]:= (*The names "jηBarMin", "jηBarMax" are similar to quantities below,
so save the current values labeled by "Best".*)
(* jηBar entries: 1. grid point # , 2. alignment angle .*)
{jηBarMinBest, jηBarMaxBest} = {jηBarMin, jηBarMax} ;
```

```
In[]:= Print["The min alignment angle is ηmin = ", jηBarMinBest[[2]] * (360. / (2. π)),
  "° , which has a significance of sig. = ", sigηBarMin, ",  plus/minus = + ",
  sigBigηBarMin - sigηBarMin, " and - ", sigηBarMin - sigSmallηBarMin,
  " , giving a range from sig. = ", sigSmallηBarMin, " to ", sigBigηBarMin, " ."]
Print["The max avoidance angle is ηmax = ", jηBarMaxBest[[2]] * (360. / (2. π)),
  "° , which has a significance of sig. = ", sigηBarMax, ",  plus/minus = + ",
  sigBigηBarMax - sigηBarMax, " and - ", sigηBarMax - sigSmallηBarMax,
  " , giving a range from sig. = ", sigSmallηBarMax, " to ", sigBigηBarMax, " ."]
Print["These uncertainties are due to the uncertainties in the constants ci, ai."]
```

```
The min alignment angle is ηmin = 21.1667° , which has a significance of sig. =
 0.0000116577,  plus/minus = + 0.000030202 and - 9.01817×10^-6
  , giving a range from sig. = 2.6395×10^-6 to 0.0000418597 .
```

```
The max avoidance angle is ηmax = 66.6554
 ° , which has a significance of sig. = 0.000195551,  plus/minus = + 0.000358332
  and - 0.00013804 , giving a range from sig. = 0.0000575103 to 0.000553883 .
```

These uncertainties are due to the uncertainties in the constants c_i, a_i.

5b. Plot of the Alignment Angle Function $\overline{\eta}$(H)

Definitions

αjδjηBarHjTable	{α_j, δ_j, $\overline{\eta}$(H)} at each grid point $H = H_j$, in degrees
ηBarHjSmooth	interpolation of αjδjηBarHjTable yields $\overline{\eta}$(H) as a smooth function of the (α,δ) of H
xyηBarAitoffTable	{x, y, $\overline{\eta}$(x,y)} , where x,y are Aitoff coordinates and $\overline{\eta}$(x,y) is the alignment angle
xyAitoffSources	{x,y} Aitoff coordinates for the sources' locations on the sphere
dηContourPlot	separation of successive contour lines, in degrees
listCP	list contour plot of $\overline{\eta}$(H) from xyηBarAitoffTable
mapOfηBar	contour plot of the alignment angle $\overline{\eta}$(H) , adorned with source locations and labels
rCenterSrc	arithmetic average of the radial unit vectors to the sources, previously called sourceCenter
rHmin, rHmax	radial unit vectors to the alignment and avoidance hubs $H_{\min}$ and $H_{\max}$
rPerpHmin (max)	a unit vector in the plane of the great circle combining rCenterSrc and rHmin (max)
rGreatMinCircle(θ) (Max)	radial unit vector to a point on the great circle
αGreatMin (Max)	longitude at the point for θ
δGreatMin (Max)	latitude at the point for θ
xyAitoffGreatMin (Max)	Aitoff plot coordinates for the great circles
crossMin (Max)	unit vector perpendicular, normal to the plane of the great circle
θminMAXgreatcircles	angle between the vectors normal to the planes of the two great circles

In[∘]:=
```
rCenterSrc0 = (1/nSrc) Sum[rSrc[[i]], {i, Length[rSrc]}];
rCenterSrc = rCenterSrc0 / (rCenterSrc0.rCenterSrc0)^(1/2.);
```

In[∘]:=
```
rHmin = er[ αHminDegrees (2. π/360.) + π, -δHminDegrees (2. π/360.) ];
rPerpHmin0 = rHmin - (rHmin.rCenterSrc) rCenterSrc;
rPerpHmin = rPerpHmin0 / (rPerpHmin0.rPerpHmin0)^(1/2.);
rGreatMinCircle[θ_] := Cos[θ] rCenterSrc + Sin[θ] rPerpHmin
αGreatMin[θ_] := αFROMr[rGreatMinCircle[θ]]
δGreatMin[θ_] := δFROMr[rGreatMinCircle[θ]]
xyAitoffGreatMin = Table[{xH180[ αGreatMin[θ] (360/(2 π)), δGreatMin[θ] (360/(2 π)) ],
    yH180[ αGreatMin[θ] (360/(2 π)), δGreatMin[θ] (360/(2 π)) ]}, {θ, 1, 360}];
```

In[•]:=
```
(rHmin.rCenterSrc);
Print["The angle between the sample's center and the alignment hub Hmin is ",
 ArcCos[-(rHmin.rCenterSrc)] (360./(2. π)), "°."]
```

The angle between the sample's center and the alignment hub H_{min} is 13.904°.

In[•]:=
```
rHmax = er[ αHmaxDegrees (2. π/360.) + π, -δHmaxDegrees (2. π/360.) ];
rPerpHmax0 = rHmax - (rHmax.rCenterSrc) rCenterSrc;
rPerpHmax = rPerpHmax0/(rPerpHmax0.rPerpHmax0)^(1/2.);
rGreatMaxCircle[θ_] := Cos[θ] rCenterSrc + Sin[θ] rPerpHmax
αGreatMax[θ_] := αFROMr[rGreatMaxCircle[θ]]
δGreatMax[θ_] := δFROMr[rGreatMaxCircle[θ]]
xyAitoffGreatMax = Table[{xH180[ αGreatMax[θ] (360/(2 π)), δGreatMax[θ] (360/(2 π)) ],
    yH180[ αGreatMax[θ] (360/(2 π)), δGreatMax[θ] (360/(2 π)) ]}, {θ, 1, 360}];
```

In[•]:=
```
(rHmax.rCenterSrc);
Print["The angle between the sample's center and the avoidance hub Hmax is ",
 ArcCos[(rHmax.rCenterSrc)] (360./(2. π)), "°."]
```

The angle between the sample's center and the avoidance hub H_{max} is 57.0234°.

In[•]:=
```
crossMin0 = Cross[rHmin, rCenterSrc];
crossMin = crossMin0/(crossMin0.crossMin0)^(1/2.);
crossMax0 = Cross[rHmax, rCenterSrc];
crossMax = crossMax0/(crossMax0.crossMax0)^(1/2.);
θminMAXgreatcircles = ArcCos[crossMax.crossMin] (360./(2. π));
```

In[•]:=
```
(*The following table αjδjηBarHjTable is created to be interpolated below,
yielding a smooth function ηBarHjSmooth of the alignment angle η̄(H) over the sphere.*)
(* Table αjδjηBarHjTable
 entries: 1.  α  2. δ  3. alignment angle ηBarRgnkj at grid point (all in degrees)*)
αjδjηBarHjTable = ( αjδjηBarHjTable0 = {};
   For[j = 1, j ≤ Length[jηBarHj], j++,
    AppendTo[ αjδjηBarHjTable0,  {αGrid[[j]] * (360. / (2. π)), δGrid[[j]] * (360. / (2. π)),
      jηBarHj[[j, 2]] * (360. / (2. π))}]  ; If[ 360. ≥ αGrid[[j]] * (360. / (2. π)) > 354.,
     AppendTo[ αjδjηBarHjTable0,  {αGrid[[j]] * (360. / (2. π)) - 360.,
       δGrid[[j]] * (360. / (2. π)), jηBarHj[[j, 2]] * (360. / (2. π))}]  ] ;
    If[ +6. > αGrid[[j]] * (360. / (2. π)) ≥ 0., AppendTo[ αjδjηBarHjTable0,  {αGrid[[j]] * (360. / (2.
            π)) + 360, δGrid[[j]] * (360. / (2. π)), jηBarHj[[j, 2]] * (360. / (2. π))}]  ]        ];
   αjδjηBarHjTable0);
```

In[ ]:=
```
ηBarHjSmooth = Interpolation[αjδjηBarHjTable] (*The smooth alignment angle function η̄(H).*)
```

••• Interpolation: Interpolation on unstructured grids is currently only supported for InterpolationOrder->1 or InterpolationOrder->All. Order will be reduced to 1.

Out[ ]=
```
InterpolatingFunction[ Domain: {{-5.92, 366.}, {-88., 88.}} Output: scalar ]
```

In[ ]:=
```
(*Transcribe the alignment function  η̄(H), the location of the sources,
and the Celestial Equator onto an Aitoff plot.*)
xyηBarAitoffTable = Partition[Flatten[Table[
    {xH180[α, δ], yH180[α, δ], ηBarHjSmooth[α, δ]}, {α, 2., 358., 2.}, {δ, -88., 88., 2.}]], 3];
(* The smooth alignment angle function η̄(H) = ηBarHjSmooth mapped
  onto a 2D Aitoff projection of the sphere. *)

xyAitoffSources = Table[{xH180[ αSrc[[n]] (360 / (2 π)), δSrc[[n]] (360 / (2 π)) ],
    yH180[ αSrc[[n]] (360 / (2 π)), δSrc[[n]] (360 / (2 π)) ]}, {n, nSrc}];
 (*The Aitoff coordinates for the sources' locations.*)
```

In[ ]:=
```
xH180[0, 0]
```

Out[ ]=
```
-3.14159
```

In[ ]:=
```
(* Contour plot of the alignment function ηBarHjSmooth. *)
dηContourPlot = 5 ;
(*, in degrees. *)listCP = ListContourPlot[Union[xyηBarAitoffTable(*,{{xH180[αHminDegrees,
      δHminDegrees],yH180[αHminDegrees,δHminDegrees],ηBarMin*(360./(2.π))-1.0}},
   {{xH180[αHmaxDegrees,δHmaxDegrees],yH180[αHmaxDegrees,δHmaxDegrees],ηBarMax*(360./(2.π))+
      1.0}}*)], AspectRatio → 1 / 2, Contours → Table[η, {η, Floor[jηBarMin[[2]] * (360. / (2. π))] +
      1, Ceiling[jηBarMax[[2]] * (360. / (2. π))] - 1, dηContourPlot}],
  ColorFunction → "TemperatureMap", PlotRange → {{-5.5, 5.5}, {-3, 3}}, Axes -> False,
  Frame → False, (*PlotLabel→"The alignment function η̄(H)",*)PlotLegends → Automatic] ;
```

```
In[]:= (*Construct the map of η̄(H).*)
mapOfηBar =
  Show[{listCP, Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]},
      {δ, -90, 90}, PlotStyle → {Black, Thickness[0.002]}, (*Mesh→{11,5,0}
      (*{23,11,0}*),MeshStyle→Thick,*)PlotPoints → 60], {α, 0, 360, 30}], Table[
     ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, 0, 360}, PlotStyle → {Black, Thickness[0.002]},
      (*Mesh→{11,5,0}(*{23,11,0}*),MeshStyle→Thick,*)PlotPoints → 60], {δ, -60, 60, 30}],
    Graphics[{PointSize[0.004], Text[StyleForm["N", FontSize -> 14, FontWeight -> "Plain"],
       {0, 1.85}], (*Sources S:*)Green, Point[ xyAitoffSources ], Gray,
      PointSize[0.002], Point[ xyAitoffGreatMin ], Point[ xyAitoffGreatMax ], Black,
      Text[StyleForm["Hmax", FontSize → 12, FontWeight -> "Bold"], {-3.3, -1.0}],
      {Arrow[BezierCurve[{{-3.3, -1.2}, {-1.3, -3.0}, {xH180[αHmaxDegrees - 180, -δHmaxDegrees],
           yH180[αHmaxDegrees - 180, -δHmaxDegrees]}}]]},
      Text[StyleForm["Hmin", FontSize → 12, FontWeight -> "Bold"], {3.3, -1.0}],
      {Arrow[BezierCurve[{{3.3, -1.2}, {0.3, -3.0},
          {xH180[αHminDegrees, δHminDegrees], yH180[αHminDegrees, δHminDegrees]}}]]},
      Text[StyleForm["Hmin", FontSize → 12, FontWeight -> "Bold"], {-3.3, 1.0}],
      {Arrow[BezierCurve[{{-3.3, 1.2}, {-2.3, 2.0}, {xH180[αHminDegrees - 180, -δHminDegrees],
           yH180[αHminDegrees - 180, -δHminDegrees]}}]]}, (**)
      Text[StyleForm["Hmax", FontSize → 12, FontWeight -> "Bold"], {3.3, 1.0}] ,
      {Arrow[BezierCurve[{{3.3, 1.2}, {2.3, 2.0},
          {xH180[αHmaxDegrees, δHmaxDegrees], yH180[αHmaxDegrees, δHmaxDegrees]}}]]}
         }]}, ImageSize → 1.5×432];
```

5c. Section Summary

This sample is an extreme case, the alignment hub $H_{\min}$ is very close to the sources.

We include the Great Circle from the center of the sources to the alignment hub $H_{\min}$ on the map. We also draw the Great Circle from source center to the avoidance hub $H_{\max}$. The two Great Circles divide the sphere quite evenly, the two Great Circles are perpendicular at the two points where they cross, within experimental error.

In[ ]:=
```
mapOfηBar
Print[
 "Figure 5: The alignment function η̅(H), Eq. (1). The map is centered on (α,δ)=(180°,0°),"]
Print["with α = 0° on the left and α = 360° on the right, Equatorial Coordinates."]
Print["The sources are located at the dots, shaded ", Green, " ."]
Print["The smallest alignment angle is η̅min = ",
 Round[jηBarMinBest[[2]] (360. / (2. π))], "°, located at the"]
Print["alignment hubs Hmin and -Hmin in the areas shaded ", Blue, " . "]
Print["The hubs Hmin and -Hmin are located at (α,δ) = ", Round[{αHminDegrees, δHminDegrees}],
 " and ", Round[{αHminDegrees - 180, -δHminDegrees}], " , in degrees."]
Print["The largest avoidance angle is η̅max = ",
 Round[jηBarMaxBest[[2]] (360. / (2. π))], "°, located at the"]
Print["avoidance hubs Hmax and -Hmax in the areas shaded ", Red, " . "]
Print["The hubs Hmax and -Hmax are located at (α,δ) = ",
 Round[{αHmaxDegrees - 180, -δHmaxDegrees}], " and at ",
 Round[{αHmaxDegrees, δHmaxDegrees}], " , in degrees."]
Print["To guide the eye, two Great Circles are plotted, one through the sources' center and the
   avoidance hubs Hmax and -Hmax. The other connects the center of the sources' locations
   with the alignment hubs Hmin and -Hmin. The Great Circles are shaded Gray, ", Gray, " ."]
Print["Notes: Although somewhat obscured by the distortion needed to plot a
   sphere on a flat surface, the function η̅(H) is symmetric across diameters.
   Diametrically opposite points -H and H have the same alignment angle η̅(H)."]
```

Out[ ]=



Figure 5: The alignment function $\overline{\eta}(H)$, Eq. (1). The map is centered on $(\alpha,\delta) = (180°,0°)$,

with α = 0° on the left and α = 360° on the right, Equatorial Coordinates.

The sources are located at the dots, shaded ■ .

The smallest alignment angle is $\overline{\eta}_{min}$ = 21°, located at the

alignment hubs H_{min} and $-H_{min}$ in the areas shaded ■ .

The hubs H_{min} and $-H_{min}$ are located at (α,δ) = {188, 0} and {8, 0} , in degrees.

The largest avoidance angle is $\overline{\eta}_{max}$ = 67°, located at the

avoidance hubs H_{max} and $-H_{max}$ in the areas shaded ■ .

The hubs H_{max} and $-H_{max}$ are located at (α,δ) = {137, -32} and at {317, 32} , in degrees.

To guide the eye, two Great Circles are plotted, one through the sources' center and the avoidance hubs H_{max} and $-H_{max}$. The other connects the center of the sources' locations with the alignment hubs H_{min} and $-H_{min}$. The Great Circles are shaded Gray, ■ .

Notes: Although somewhat obscured by the distortion needed to plot a sphere on a flat surface, the function $\overline{\eta}(H)$ is symmetric across diameters. Diametrically opposite points -H and H have the same alignment angle $\overline{\eta}(H)$.

```
(*
SetDirectory[
  "C:\\Users\\shurt\\Dropbox\\HOME_DESKTOP-0MRE5OJ\\SendXXX_CJP_CEJPetc\\SendViXra\\
    20200715AlignmentMethod\\20210505AlignmentMethodv4\\20210515Clump1QSOsNearNGP"]
 Export["20210424QSOnearbyHmin.pdf",mapOfηBar]
*)
```

In[•]:=
```
(*Statistics*)
Print["Statistics of the Alignment Function η̄(H) :"]
Print[" "]
Print["The number of sources: N = ", nSrc]
Print["The min alignment angle, ηmin = ", jηBarMinBest[[2]] * (360. / (2. π)),
 "°, is ", (η0MIN[nSrc, c1MIN, a1MIN] - jηBarMinBest[[2]]) * (360. / (2. π)),
 "° below the most likely value, ",
 η0MIN[nSrc, c1MIN, a1MIN] * (360. / (2. π)), "°, for random runs."]
Print["Since the uncertainty σ is ", σMIN[nSrc, c2MIN, a2MIN] * (360. / (2. π)),
 "°, the difference ", (η0MIN[nSrc, c1MIN, a1MIN] - jηBarMinBest[[2]]) * (360. / (2. π)),
 "° is ", (η0MIN[nSrc, c1MIN, a1MIN] - jηBarMinBest[[2]]) / σMIN[nSrc, c2MIN, a2MIN],
 "σs from the most likely random run value."]
Print["Thus, the smallest alignment angle η̄min is " ,
 (η0MIN[nSrc, c1MIN, a1MIN] - jηBarMinBest[[2]]) / σMIN[nSrc, c2MIN, a2MIN],
 "σs below the most likely random run value."]
Print[""]
Print["The largest avoidance angle, ηmax = ", jηBarMaxBest[[2]] * (360. / (2. π)),
 "°, is ", - (η0MAX[nSrc, c1MAX, a1MAX] - jηBarMaxBest[[2]]) * (360. / (2. π)),
 "° above the most likely value, ",
 η0MAX[nSrc, c1MAX, a1MAX] * (360. / (2. π)), "°, for random runs."]
Print["Since the uncertainty σ is ", σMAX[nSrc, c2MAX, a2MAX] * (360. / (2. π)),
 "°, the difference ", - (η0MAX[nSrc, c1MAX, a1MAX] - jηBarMaxBest[[2]]) * (360. / (2. π)),
 "° is ", - ((η0MAX[nSrc, c1MAX, a1MAX] - jηBarMaxBest[[2]]) / σMAX[nSrc, c2MAX, a2MAX]),
 "σs from the most likely random run value." ]
Print["Thus, the largest avoidance angle η̄max is " ,
 (jηBarMaxBest[[2]] - η0MAX[nSrc, c1MAX, a1MAX]) / σMAX[nSrc, c2MAX, a2MAX],
 "σs above the most likely random run value."]
```

```
Statistics of the Alignment Function η̄(H) :


The number of sources: N = 27

The min alignment angle, ηmin = 21.1667°, is
 14.0934° below the most likely value, 35.2602°, for random runs.

Since the uncertainty σ is 3.29664°, the difference
 14.0934° is 4.27509σs from the most likely random run value.

Thus, the smallest alignment angle η̄min is 4.27509σs below the most likely random run value.


The largest avoidance angle, ηmax = 66.6554°, is
 11.8243° above the most likely value, 54.8311°, for random runs.

Since the uncertainty σ is 3.28622°, the difference
 11.8243° is 3.59814σs from the most likely random run value.

Thus, the largest avoidance angle η̄max is 3.59814σs above the most likely random run value.
```

In[ ]:=
```
Print["The center of the sources is a point that makes a great circle, shaded ",
  Gray, " in Fig. 5, with the alignment hub Hmin."]
Print["The center of the sources makes a second great circle, shaded ",
  Gray, " in Fig. 5, with the avoidance hub Hmax."]
Print["The angle between the planes of the two great circles is ",
  θminMAXgreatcircles, "°."]
```

```
The center of the sources is a point that makes a great circle, shaded
 ■ in Fig. 5, with the alignment hub Hmin.
The center of the sources makes a second great circle, shaded
 ■ in Fig. 5, with the avoidance hub Hmax.
The angle between the planes of the two great circles is 91.1259°.
```

6. Uncertainty Runs

6a. Creating and Storing Uncertainty Runs

For each "uncertainty run", the polarization direction ψ for each source is allowed to differ from the best value ψn by an amount $\delta\psi$ chosen according to a Gaussian distribution with mean (best) value ψn and half-width $\sigma\psi$, $\psi = \psi$n + $\delta\psi$. Both values ψn and $\sigma\psi$ are taken from the catalogs.

Definitions:

rSrcxrGrid unit vector $S_i \times H_j$ in the direction of the cross product of the radial vector S_i to a source with the radial vector H_j to a grid point

μ the mean value μ of the measurement Gaussian for ψ

σ the uncertainty of the measured polarization position angle ψ

ψData polarization directions $\psi = \psi$n + $\delta\psi$

runData collection of data to save from the uncertainty runs, see below for content list

nRunPrint dummy index controlling when current TimeUsed and MemoryInUse are printed

ψSrc the polarization direction ψ for the run.

rSrcxψSrc unit vector, $S_i \times \psi_i$, cross product of the radial vector S_i to the source with the vector $\hat{v}_\psi$ in the direction of the polarization

jηBarToGrid {j, $\overline{\eta}(H_j)$}, where j is the index for the grid point H_j and $\overline{\eta}(H_j)$ is the alignment angle function, (1), at H_j

sortjηBarToGrid sort {j, $\overline{\eta}(H_j)$}, with the smaller angle $\overline{\eta}$(H) first.

jηBarMin1 {j,$\overline{\eta}$(H)} for the smallest value of $\overline{\eta}$(H) , best alignment

jηBarMax1 {j,$\overline{\eta}$(H)}, for the largest value of $\overline{\eta}$(H) , most avoided

ηBarMinData values of $\overline{\eta}_{min}$ from uncertainty runs, alignment

ηBarMaxData values of $\overline{\eta}_{max}$ from uncertainty runs, avoidance

HminαData values of $\alpha = \alpha$ for hub H_{min} from uncertainty runs, alignment

HminδData values of $\delta = \delta$ for hub H_{min} from uncertainty runs, alignment

HmaxαData values of $\alpha = \alpha$ for hub H_{max} from uncertainty runs, avoidance

HmaxδData values of $\delta = \delta$ for hub H_{max} from uncertainty runs, avoidance

Tables:

ψData	entries: 1. Run #	2. ψSrc, list of polarization position angles ψ
runData	entries: 1. Run #	2. $\{\overline{\eta}_{\min}, \{\alpha,\delta\}$ at $H_{\min}\}$ 3. $\{\overline{\eta}_{\max}, \{\alpha,\delta\}$ at $H_{\max}\}$

To create Uncertainty Runs, first calculate "rSrcxrGrid" and then evaluate the "For" statement in the following two cells. One can save the results with the "Put[]" statements.

Once saved, there is no need to repeat the runs. Comment out the "rSrcxrGrid" and "For" statements by enclosing each in (*comment brackets*). The data can be retrieved with the "Get" statements.

```
rSrcxrGrid1 = Table[ Cross[ rSrc[[i]], rGrid[[j]] ] ,   {i, nSrc}, {j, nGrid}];
(*first step: αw cross product, not unit vectors*)
rSrcxrGrid = Table[ rSrcxrGrid1[[i, j]] /
    (rSrcxrGrid1[[i, j]].rSrcxrGrid1[[i, j]] + 0.000001)^1/2. ,   {i, nSrc}, {j, nGrid}];
Clear[rSrcxrGrid1];
```

```
(*rSrcxrGrid: table of the unit vectors perpendicular to the plane
  of the great circle containing the source Si and the grid point Hj*)
```

```
nR = 5000;(*number of runs with the PPA ψ allowed by measurement uncertainty. *)
μ = ψn; σ = σψn; runData = {}; ψData = {}; nRunPrint = 0;
For[nRun = 1, nRun ≤ nR, nRun++,
 If[nRun > nRunPrint, Print["At the start of run ", nRun, ", the time is ",
   TimeUsed[], " seconds and the memory in use is ", MemoryInUse[], " bytes."];
  nRunPrint = nRunPrint + 500];
  ψSrc = Table[RandomVariate[NormalDistribution[μ[[i]], σ[[i]]]], {i, nSrc}];
 (*table of PPA angles ψ for the sources in region j0, in radians*)
 rSrcxψSrc  = Table[ Sin[ψSrc[[i]]] eNSrc[[i]] - Cos[ψSrc[[i]]] eESrc[[i]],   {i, nSrc}];
 (*table of the cross product of rSrc and vector in direction of ψSrc,
 a unit vector*)jηBarToGrid  = Table[{j, (1/nSrc) Sum[ArcCos[
       Abs[ rSrcxψSrc[[i]].rSrcxrGrid[[i, j]] ] - 0.000001 ], {i, nSrc}]}, {j, nGrid}];
 (*
 {grid point #, value of the alignment angle ηnHj[j] averaged over all sources,
  in radians}*) sortjηBarToGrid = Sort[jηBarToGrid, #1[[2]] < #2[[2]] &];
 (*jηBarToGrid, {j,ηj}, but sorted with the smallest alignment angles first
 *)
 jηBarMin1 = sortjηBarToGrid[[1]]; (*  {j,ηj}, at the grid point Hj with minimum η̄*)
 jηBarMax1 = sortjηBarToGrid[[-1]]; (*  {j,ηj},
 at the grid point Hj with maximum η̄*)AppendTo[ψData, {nRun, ψSrc}];
 AppendTo[runData, {nRun, { jηBarMin1[[2]],
    {αGrid [ [ jηBarMin1[[1]] ]], δGrid [[ jηBarMin1[[1]] ]]}}, { jηBarMax1[[2]],
    {αGrid [[ jηBarMax1[[1]] ]], δGrid [[ jηBarMax1[[1]] ]]}}} ](*collect data*)  ]
```

```
At the start of run 1, the time is 13.39 seconds and the memory in use is 214 284 616 bytes.
At the start of run 501, the time is 377.109 seconds and the memory in use is 231 208 232 bytes.
At the start of run 1001, the time is 743.218 seconds and the memory in use is 231 748 936 bytes.
At the start of run 1501, the time is 1096.95 seconds and the memory in use is 232 297 288 bytes.
At the start of run 2001, the time is 1453.58 seconds and the memory in use is 232 846 056 bytes.
At the start of run 2501, the time is 1810.16 seconds and the memory in use is 233 394 888 bytes.
At the start of run 3001, the time is 2160.5 seconds and the memory in use is 233 943 656 bytes.
At the start of run 3501, the time is 2513.25 seconds and the memory in use is 234 492 296 bytes.
At the start of run 4001, the time is 2865.47 seconds and the memory in use is 235 041 128 bytes.
At the start of run 4501, the time is 3217.08 seconds and the memory in use is 235 589 896 bytes.
```

Hint: You can save memory if you do not get the “ψData”. The table ψData is needed to reconstruct the exact values of the runData table, but it is not needed in any following calculation.

```
SetDirectory[homeDirectory];(*Save memory space; ψData is not used below.*)
(*
Put[ψData,"20210509PsiDataClump1RA175Dec10.dat"  ]  (*Save a new "ψData"*)
*)
(*ψData=Get["20210509PsiDataClump1RA175Dec10.dat"]; *) (*Get an old "ψData"*)
```

Hint: Saving “runData” to a file avoids the time it takes to complete the “For” statement. Make the above “For” statement into a remark so that it doesn’t evaluate.

```
 SetDirectory[homeDirectory];
(*
Put[runData,"20210509runDataClump1RA175Dec10.dat"  ]  (*Save a new "runData".*)
*)
(*
runData=Get["20210509runDataClump1RA175Dec10.dat"];
*) (*Get an old "runData".*)
```

```
In[∘]:= Print["The number of uncertainty runs is ", Length[runData], "."]
```

```
The number of uncertainty runs is 5000.
```

```
In[∘]:= ηBarMinData = Table[runData[[i1, 2, 1]] , {i1, Length[runData]}];
ηBarMaxData = Table[runData[[i1, 3, 1]] , {i1, Length[runData]}];
HminαData = Table[ runData[[i1, 2, 2, 1]] , {i1, Length[runData]}];
HminδData = Table[runData[[i1, 2, 2, 2]], {i1, Length[runData]}];
HmaxαData = Table[ runData[[i1, 3, 2, 1]] , {i1, Length[runData]}];
HmaxδData = Table[runData[[i1, 3, 2, 2]], {i1, Length[runData]}];
```

6b. The Effects of Uncertainty on the Smallest Alignment Angle $\overline{\eta}_{\min}$

This section fits a Gaussian distribution to the $\overline{\eta}_{\min}$ from the uncertainty runs.

Definitions

sortηBarMin   sort the list of $\overline{\eta}_{\min}$ from the uncertainty runs

η0B	estimated mean of the Gaussian fit
σB	estimated half-width of the Gaussian fit
histogramrange	{min η, max η, Δη} for the histogram
hl0, hl	histogram {η, bin height} tables needed to set up the NonlinearModelFit
nlmB	non-linear model fit of a Gaussian to the $\overline{\eta}_{min}$ histogram
showNLMB	plot of Gaussian and histogram
ParametersNLMB	amplitude, half-width, and mean of the Gaussian fit
pTableNLMB	table of parameter attributes, including standard error

In[◦]:=
```
sortηBarMin = Sort[ηBarMinData];
η0B = mean[ηBarMinData ]; (*Guess the mean for the Gaussian. *)
σB = stanDev[ηBarMinData ];(*Guess the half-width.*)
histogramrange = {η0B - 5 σB, η0B + 5 σB, 0.4 σB};
hl0 = HistogramList[sortηBarMin, histogramrange];
hl =
 Table[{(1/2) (hl0[[1, i1]] + hl0[[1, i1 + 1]]), hl0[[2, i1]]}, {i1, Length[ hl0[[2]] ]}];
nlmB = NonlinearModelFit[hl, a Exp[-(1/2.) ((x - x0) / b)^2],
  {{a, Length[sortηBarMin / 6]}, {b, σB}, {x0, η0B}}, x];(*x is ηBarMin*)
```

In[◦]:=
```
showNLMB = Show[{Histogram[sortηBarMin, histogramrange,
    PlotLabel → "η̄min ", AxesLabel → {"η̄min, radians", "ΔR"}],
   Plot[Normal[nlmB], {x, η0B - 5 σB, η0B + 5 σB}, PlotLabel → "η̄min"],
   ListPlot[hl, PlotLabel → "η̄min"] }]
Print["Figure 6: The Gaussian fit to the alignment angle
   η̄min histogram, where the height is the number "]
Print["of runs ΔR in each bin of width Δη̄min  =  ", 0.4 σB, " radians. "]
Print["The total number of runs is R = Σ(ΔR) = ", Length[runData], "."]
```

Out[◦]=



Figure 6: The Gaussian fit to the alignment angle $\overline{\eta}_{min}$ histogram, where the height is the number

of runs ΔR in each bin of width $\Delta\overline{\eta}_{min}$ = 0.00596687 radians.

The total number of runs is R = Σ(ΔR) = 5000.

In[∘]:=
```
ParametersNLMB = {a, b, x0} /. nlmB["BestFitParameters"];
pTableNLMB = nlmB["ParameterTable"]
{σηBarMinFit, ηBarMinFit} = {ParametersNLMB[[2]] , ParametersNLMB[[3]] };(*radians*)
```

Out[∘]=

	Estimate	Standard Error	t-Statistic	P-Value
a	799.527	10.5054	76.1063	3.83557×10^{-28}
b	0.0148443	0.00022522	65.91	8.9616×10^{-27}
x0	0.37913	0.00022522	1683.38	1.03855×10^{-57}

6c. The Effects of Uncertainty on the Largest Avoidance Angle $\overline{\eta}_{\max}$

This section fits a Gaussian distribution to the $\overline{\eta}_{\max}$ returned by the uncertainty runs.

Definitions: Check the list of Definitions in Sec. 6b. Trade avoidance (Max) here for alignment (Min) there.

In[∘]:=
```
sortηBarMax = Sort[ηBarMaxData];
η0MaxB = mean[ηBarMaxData ]; (*Guess the mean for the Gaussian. *)
σMaxB  = stanDev[ηBarMaxData ];(*Guess the half-width.*)
histogramrangeMAX = {η0MaxB - 5 σMaxB, η0MaxB + 5 σMaxB, 0.4 σMaxB};
hl0Max = HistogramList[sortηBarMax, histogramrangeMAX];
hlMax = Table[{(1/2) (hl0Max[[1, i1]] + hl0Max[[1, i1 + 1]]), hl0Max[[2, i1]]},
   {i1, Length[ hl0Max[[2]] ]}];
nlmMaxB = NonlinearModelFit[hlMax, a Exp[-(1/2.) ((x - x0) / b)^2],
   {{a, 300.}, {b, σMaxB}, {x0, η0MaxB}}, x];(*x is ηBarMax *)
```

In[∘]:=
```
showNLMMaxB = Show[{Histogram[sortηBarMax,
     histogramrangeMAX, PlotLabel → "η̄max", AxesLabel → {"η̄max, radians", "ΔR"}],
    Plot[Normal[nlmMaxB], {x, η0MaxB - 5 σMaxB, η0MaxB + 5 σMaxB}, PlotLabel → "η̄max"] ,
    ListPlot[hlMax, PlotLabel → "η̄max"]}]
Print["Figure 7: The Gaussian fit to the avoidance angle η̄max
    histogram. The bins have a width Δη̄max  =  ", 0.4 σMaxB,
  " radians and have a height equal to the number of runs ΔR in the bin."]
Print["The total number of runs is R = Σ(ΔR) = ", Length[runData], "."]
```

Out[∘]=



Figure 7: The Gaussian fit to the avoidance angle $\overline{\eta}_{\max}$ histogram. The bins have a width $\Delta\overline{\eta}_{\max}$ = 0.00645648 radians and have a height equal to the number of runs ΔR in the bin.

The total number of runs is R = Σ(ΔR) = 5000.

In[ ]:=
```
ParametersNLMMaxB = {a, b, x0} /. nlmMaxB["BestFitParameters"];
pTableNLMMaxB = nlmMaxB["ParameterTable"]
{σηBarMaxFit, ηBarMaxFit} = {ParametersNLMMaxB[[2]] , ParametersNLMMaxB[[3]] };
(*radians*)
```

Out[ ]=

	Estimate	Standard Error	t-Statistic	P-Value
a	799.002	8.28405	96.4506	2.12287×10^{-30}
b	0.0160853	0.000192572	83.5287	4.98483×10^{-29}
x0	1.15318	0.000192572	5988.32	7.79346×10^{-70}

6d. The Effects of Uncertainty on the Locations (α,δ) of the Alignment Hubs $H_{\min}$

Each uncertainty run returns an alignment hub $H_{\min}$. In this section, we calculate the mean and standard deviation to approximate the distribution of the locations the Alignment Hubs $H_{\min}$.

In any one run, the analysis produces an alignment angle $\overline{\eta}$ at each grid point. There can be just one minimum alignment angle $\overline{\eta}_{\min}$, but there are two hubs, $H_{\min}$ and $-H_{\min}$, by the symmetry across a diameter. So we collect all the hubs together by moving the $-H_{\min}$ hubs across a diameter to join the $H_{\min}$ hubs.

Definitions

Hminα — α in radians for $H_{\min}$
Hminδ — δ in radians for $H_{\min}$
$\sigma\alpha$MinFit1 — half-width for α uncertainty runs
αMinFit1mean for α uncertainty runs
$\sigma\delta$MinFit1 — half-width for δ uncertainty runs
δMinFit1 — mean for δ uncertainty runs
HminαAVE — average over all uncertainty runs of α for $H_{\min}$
Hmin$\alpha\delta$ — (α,δ) table for ListPlot
lpHmin — plot Hmin hubs from uncertainty runs
α1,2Min1 — values needed for framing the most likely hubs
δ1,2Min1 — ditto for latitude

In[ ]:=
```
(* Gather the hubs. Move the hubs across diameters,
Δα = π, or around a complete circle, Δα = 360°,
if necessary, so that all hubs satisfy 0° ≤  α  <  180° .*)
Hminα0 = HminαData;
Hminδ0 = HminδData;
HminαBy180n = Round[Hminα0 / π];
Hminα1 = Table[Hminα0[[i1]] - HminαBy180n[[i1]] π , {i1, Length[Hminα0]}];
Hminδ1 = Table[(-1)^HminαBy180n[[i1]] Hminδ0[[i1]] , {i1, Length[Hminδ0]}];
Hminα = Table[
   If[Hminα1[[i1]] < 0, Hminα1[[i1]] + π, Hminα1[[i1]], "huh?"] , {i1, Length[Hminα1]}];
Hminδ = Table[If[Hminα1[[i1]] < 0, -Hminδ1[[i1]], Hminδ1[[i1]], "huh?"] ,
   {i1, Length[Hminδ1]}];
```

In[∘]:=
```
(*Check that 0° ≤  α  <  180°  and -90° ≤  δ  <  90° *)
(*ListPlot[{Sort[Hminα],Sort[Hminδ]},
  PlotLabel→"α and δ for Hmin, radians",AxesLabel→{"Run #","α,δ"}]
{Sort[Hminα][[1]],Sort[Hminα][[-1]]}(360./2.π)(*degrees*)
 {Sort[Hminδ][[1]],Sort[Hminδ][[-1]]}(360./2.π)  (*degrees*)
*)
```

In[∘]:=
```
{σαMinFit1, αMinFit1} = {stanDev[Hminα], mean[Hminα]};(*radians*)
{σδMinFit1, δMinFit1} = {stanDev[Hminδ], mean[Hminδ]};(*radians*)
```

In[∘]:=
```
(*Define quantities for the plot of the Hmin from the uncertainty runs. *)
Hminαδ = Sort[Table[{Hminα[[i5]], Hminδ[[i5]]}, {i5, Length[Hminα]}]];
{Hminαδ[[1]], Hminαδ[[-1]]} ;(*radians*)
{Hminαδ[[1]], Hminαδ[[-1]]} (360. / (2. π)) ;(*degrees*)
lpHmin = ListPlot[Hminαδ (360. / (2. π)),
   PlotRange → {{0, 360}, {-90, 90}}, PlotMarkers → Automatic,
   AxesLabel → {"α, degrees", "δ, degrees"}, PlotLabel → "(α,δ) for the Hmin hubs",
   Ticks → {Table[{t, t}, {t, 0, 360, 45}], Automatic}];
α1Min1 = (αMinFit1 - σαMinFit1) (360. / (2. π));
α2Min1 = (αMinFit1 + σαMinFit1) (360. / (2. π));
δ1Min1 = (δMinFit1 - σδMinFit1) (360. / (2. π));
δ2Min1 = (δMinFit1 + σδMinFit1) (360. / (2. π));
```

6e. The Effects of Uncertainty on the Locations (α,δ) of the Avoidance Hubs $H_{\max}$.

Each uncertainty run returns an alignment hub $H_{\max}$. In this section, we calculate the mean and standard deviation all such hubs to approximate the distribution of the locations of the Avoidance Hubs $H_{\max}$.

Definitions: Explore the definitions for $H_{\min}$ at the start of Sec. 6d. Find the similarly named quantity by interchanging Max for Min. Adjust the definition to the present context.

In[∘]:=
```
(* Move hubs, if necessary, so that 0°  ≤  α  <  360° *)
Hmaxα0 = HmaxαData;
Hmaxδ0 = HmaxδData;
HmaxαBy180n = Round[Hmaxα0 / π];
Hmaxα1 = Table[Hmaxα0[[i1]] - HmaxαBy180n[[i1]] π , {i1, Length[Hmaxα0]}];
Hmaxδ1 = Table[ (-1)^HmaxαBy180n[[i1]] Hmaxδ0[[i1]] , {i1, Length[Hmaxδ0]}];
Hmaxα = Table[
   If[0 > Hmaxα1[[i1]], Hmaxα1[[i1]] + π, Hmaxα1[[i1]], "huh?"] , {i1, Length[Hmaxα1]}];
Hmaxδ = Table[If[0 > Hmaxα1[[i1]], -Hmaxδ1[[i1]], Hmaxδ1[[i1]], "ah"] ,
   {i1, Length[Hmaxδ1]}];
```

In[∘]:=
```
(*Check that 0° ≤  α  <  180°  and -90° ≤  δ  <  90° *)
(*ListPlot[{Sort[Hmaxα],Sort[Hmaxδ]},PlotRange→{-2π,2π},
  AxesLabel→{"Run #","α,δ radians"},PlotLabel→"αs, δs for Hmax"]
 {Sort[Hmaxα][[1]],Sort[Hmaxα][[-1]]}(360./2.π)(*degrees*)
 {Sort[Hmaxδ][[1]],Sort[Hmaxδ][[-1]]}(360./2.π) *) (*degrees*)
```

```
In[ ]:= {σαMaxFit, αMaxFit} = {stanDev[Hmaxα], mean[Hmaxα]};(*radians*)
        {σδMaxFit, δMaxFit} = {stanDev[Hmaxδ], mean[Hmaxδ]};(*radians*)
```

```
In[ ]:= (* Define quantities for the plot of the
         locations of the Hmax from the uncertainty runs. *)
        Hmaxαδ = Table[{Hmaxα[[i8]], Hmaxδ[[i8]]}, {i8, Length[Hmaxδ ]}];
        {Hmaxαδ[[1]], Hmaxαδ[[-1]]} ;(*radians*)
        {Hmaxαδ[[1]], Hmaxαδ[[-1]]} (360. / (2. π)) ;(*degrees*)
        lpHmax1 = ListPlot[Hmaxαδ (360. / (2. π)), PlotRange → {{0, 360}, {-90, 90}},
           PlotMarkers → Automatic, AxesLabel → {"α, degrees", "δ, degrees"},
           PlotLabel → "Hmax hubs  ", Ticks → {Table[{t, t}, {t, 0, 360, 45}], Automatic}];
        α1Max = (αMaxFit - σαMaxFit) (360. / (2. π));
        α2Max = (αMaxFit + σαMaxFit) (360. / (2. π));
        δ1Max = (δMaxFit - σδMaxFit) (360. / (2. π));
        δ2Max = (δMaxFit + σδMaxFit) (360. / (2. π));
```

6f. The Effects of Uncertainty on the angle θ between the planes of the Sample to H_{min} Great Circle and the Sample to H_{max} Great Circle.

These are the Gray lines in Fig. 5.

Definitions:

"uRuns" prefix	results from the uncertainty runs
uRunsCrossMin	unit vector normal to the Great Circle connecting the center of the source region with the alignment hub H_{min}
uRunsCrossMax	unit vector normal to the Great Circle connecting the center of the source region with the alignment hub H_{max}
uRunsθminMAXgreatcircles	angle between the two normals in degrees
sortθminMAX	sort "uRunsθminMAXgreatcircles", smallest θ first

See Definitions above in Secs. 6a,6b for other quantities below. There you should find similarly named quantities.

```
In[ ]:= uRunsCrossMin0 =
          Table[Cross[er[Hminα[[i]], Hminδ[[i]]], sourceCenter ], {i, Length[Hminα]}];
        uRunsCrossMin = Table[ uRunsCrossMin0[[i]] / (uRunsCrossMin0[[i]].uRunsCrossMin0[[i]])^(1/2.),
           {i, Length[Hminα]}];
        uRunsCrossMax0 = Table[Cross[er[Hmaxα[[i]], Hmaxδ[[i]]], sourceCenter ],
           {i, Length[Hmaxα]}];
        uRunsCrossMax = Table[ uRunsCrossMax0[[i]] / (uRunsCrossMax0[[i]].uRunsCrossMax0[[i]])^(1/2.),
           {i, Length[Hmaxα]}];
        uRunsθminMAXgreatcircles = Table[ArcCos[uRunsCrossMax[[i]].uRunsCrossMin[[i]]] (360./(2. π)),
           {i, Length[Hmaxα]}];
```

In[∘]:=
```
sortΘminMAX = Sort[uRunsΘminMAXgreatcircles];
η0Θ = mean[uRunsΘminMAXgreatcircles]; (*Guess the mean for the Gaussian. *)
σΘ = stanDev[uRunsΘminMAXgreatcircles];(*Guess the half-width.*)
histogramrange = {η0Θ - 5 σΘ, η0Θ + 5 σΘ, 0.4 σΘ};
hl0 = HistogramList[sortΘminMAX, histogramrange];
hl =
 Table[{(1/2) (hl0[[1, i1]] + hl0[[1, i1 + 1]]), hl0[[2, i1]]}, {i1, Length[ hl0[[2]] ]}];
nlmΘ = NonlinearModelFit[hl, a Exp[- (1/2.) ((x - x0) / b)^2],
  {{a, Length[sortΘminMAX / 6]}, {b, σΘ}, {x0, η0Θ}}, x];(*x is ΘminMAX*)
```

In[∘]:=
```
showNLMΘ = Show[{Histogram[sortΘminMAX, histogramrange,
    PlotLabel → "Angle Θ between the Two Gray Great Circles in Fig. 5",
    AxesLabel → {"Θ, degrees", "ΔR"}],
   Plot[Normal[nlmΘ], {x, η0Θ - 5 σΘ, η0Θ + 5 σΘ}], ListPlot[hl] }]
Print["Figure 8: The Gaussian fit to the angle Θ histogram,
   where the height is the number of runs ΔR in"]
Print[" each bin of width ΔΘ  =  ", 0.4 σΘ, " degrees."]
Print[" The total number of runs is R = Σ(ΔR)  = ", Length[runData], "."]
```

Out[∘]=



Figure 8: The Gaussian fit to the angle Θ histogram, where the height is the number of runs ΔR in

each bin of width ΔΘ = 1.47323 degrees.

The total number of runs is R = Σ(ΔR) = 5000.

In[∘]:=
```
ParametersNLMΘ = {a, b, x0} /. nlmΘ["BestFitParameters"];
pTableNLMΘ = nlmΘ["ParameterTable"]
{σΘminMAXFit, ΘminMAXFit} = {ParametersNLMΘ[[2]] , ParametersNLMΘ[[3]] };(*degrees*)
```

Out[∘]=

	Estimate	Standard Error	t-Statistic	P-Value
a	783.396	26.644	29.4024	3.74098×10^{-19}
b	3.78289	0.148563	25.4632	8.12067×10^{-18}
x0	92.3208	0.148563	621.425	3.44853×10^{-48}

6g. Map of the Hubs for the Uncertainty Runs

In this subsection, we map the locations of the many alignment hubs $H_{\min}$ and the locations of the avoidance hubs $H_{\max}$ that are found in the uncertainty runs.

Definitions:

xyAitoffHmin	Aitoff coordinates for the alignment hubs H_{min} from the uncertainty runs
xyAitoffHmax	Aitoff coordinates for the avoidance hubs H_{max} from the uncertainty runs
xyAitoffOppositeHmin	Aitoff coordinates for the $-H_{min}$
xyAitoffOppositeHmax	Aitoff coordinates for the $-H_{max}$
mapOfσψHminHmax	plot of the alignment and avoidance hubs H_{min}, $-H_{min}$, H_{max}, and $-H_{max}$

In[ ]:=
```
(*The Aitoff coordinates for the hubs Hmin locations.*)
xyAitoffHmin = Table[{xH180[ Hminα [[n]] (360 / (2 π)), Hminδ[[n]] (360 / (2 π)) ],
    yH180[ Hminα [[n]] (360 / (2 π)), Hminδ[[n]] (360 / (2 π)) ]}, {n, Length[Hminδ ]}];
(*The Aitoff coordinates for the hubs Hmax locations.*)
xyAitoffHmax = Table[{xH180[ Hmaxα [[n]] (360 / (2 π)), Hmaxδ[[n]] (360 / (2 π)) ],
    yH180[ Hmaxα [[n]] (360 / (2 π)), Hmaxδ[[n]] (360 / (2 π)) ]}, {n, Length[Hminδ ]}];
(*The Aitoff coordinates for the hubs -Hmin locations.*)
xyAitoffOppositeHmin = Table[{xH180[ If[0 ≤ Hminα [[n]] (360 / (2 π)) < +180,
      Hminα [[n]] (360 / (2 π)) + 180, If[360 > Hminα [[n]] (360 / (2 π)) > 180,
       Hminα [[n]] (360 / (2 π)) - 180]], -Hminδ[[n]] (360 / (2 π)) ],
    yH180[ If[0 ≤ Hminα [[n]] (360 / (2 π)) < +180, Hminα [[n]] (360 / (2 π)) + 180,
      If[360 > Hminα [[n]] (360 / (2 π)) > 180, Hminα [[n]] (360 / (2 π)) - 180]],
     -Hminδ[[n]] (360 / (2 π)) ]}, {n, Length[Hminδ ]}];
(*The Aitoff coordinates for the hubs -Hmax locations.*)
xyAitoffOppositeHmax =
  Table[{xH180[ If[0 ≤ Hmaxα [[n]] (360 / (2 π)) < +180, Hmaxα [[n]] (360 / (2 π)) + 180,
      If[360 > Hmaxα [[n]] (360 / (2 π)) > 180, Hmaxα [[n]] (360 / (2 π)) - 180]],
     -Hmaxδ[[n]] (360 / (2 π)) ], yH180[ If[0 ≤ Hmaxα [[n]] (360 / (2 π)) < +180,
      Hmaxα [[n]] (360 / (2 π)) + 180, If[360 > Hmaxα [[n]] (360 / (2 π)) > 180,
       Hmaxα [[n]] (360 / (2 π)) - 180]], -Hmaxδ[[n]] (360 / (2 π)) ]}, {n, Length[Hmaxδ ]}];
```

In[•]:=
```
(*Construct the map of uncertainty run Hmin and Hmax hubs with ± regions indicated.*)
mapOfσψHminHmax =
  Show[{Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]},
      {δ, -90, 90}, PlotStyle → {Black, Thickness[0.002]}, PlotPoints → 60,
      PlotRange → {{-7, 7}, {-3, 3}}, Axes → False], {α, 0, 360, 30}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, 0, 360}, PlotStyle →
       {Black, Thickness[0.002]}, PlotPoints → 60], {δ, -60, 60, 30}], Graphics[{PointSize[0.007],
      Text[StyleForm["N", FontSize -> 10, FontWeight -> "Plain"], {0, 1.85}], LightBlue,
      (*Hmin:*)Point[ xyAitoffHmin ], (*-Hmin:*)Point[ xyAitoffOppositeHmin ], LightRed,
      (*Hmax:*)Point[ xyAitoffHmax ], (*-Hmax:*)Point[ xyAitoffOppositeHmax ]    }],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {δ, δ1Max, δ2Max},
      PlotStyle → {Purple, Thickness[0.002]}, PlotPoints → 60], {α, α1Max, α2Max, α2Max - α1Max}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, α1Max, α2Max},
      PlotStyle → {Purple, Thickness[0.002]}, PlotPoints → 60], {δ, δ1Max, δ2Max, δ2Max - δ1Max}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {δ, -δ2Max, -δ1Max}, PlotStyle →
       {Purple, Thickness[0.002]}, PlotPoints → 60], {α, α1Max + 180, α2Max + 180, α2Max - α1Max}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, α1Max + 180, α2Max + 180},
      PlotStyle → {Purple, Thickness[0.002]}, PlotPoints → 60], {δ, -δ2Max, -δ1Max, δ2Max - δ1Max}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {δ, -δ2Min1, -δ1Min1},
      PlotStyle → {Purple, Thickness[0.002]}, PlotPoints → 60],
     {α, α1Min1 + 180, α2Min1 + 180, α2Min1 - α1Min1}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, α1Min1 + 180, α2Min1 + 180}, PlotStyle →
       {Purple, Thickness[0.002]}, PlotPoints → 60], {δ, -δ2Min1, -δ1Min1, δ2Min1 - δ1Min1}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {δ, δ1Min1, δ2Min1}, PlotStyle →
       {Purple, Thickness[0.002]}, PlotPoints → 60], {α, α1Min1, α2Min1, α2Min1 - α1Min1}],
    Table[ParametricPlot[{xH180[α, δ], yH180[α, δ]}, {α, α1Min1, α2Min1}, PlotStyle →
       {Purple, Thickness[0.002]}, PlotPoints → 60], {δ, δ1Min1, δ2Min1, δ2Min1 - δ1Min1}](**)},
   ImageSize → 1.5 × 432, PlotLabel → "The Hubs Found from the Uncertainty Runs"];
```

6h. Section Summary

```
In[•]:= Print["To estimate the effects of experimental uncertainty, there were ",
         Length[runData], " uncertainty runs."]
        Print["Uncertainty runs have polarization directions ψ = ψn + δψ, ",
         "where δψ is chosen with a normal
           distribution of half-width σψ about the best value ψn."]
        Print["The uncertainty runs determine the smallest alignment angle to be  η̄min = ",
         ηBarMinFit (360. / (2. π)), "° ± ", σηBarMinFit (360. / (2. π)), "°."]
        Print["The uncertainty runs determine the largest avoidance angle to be  η̄max = ",
         ηBarMaxFit (360. / (2. π)), "° ± ", σηBarMaxFit (360. / (2. π)), "°."]
        Print["The uncertainty runs give the location
           for one of the alignment hub Hmin  as  (α, δ)  = ",
         {αMinFit1 (360. / (2. π)) + 180, -δMinFit1 (360. / (2. π))}, " ± ",
         {σαMinFit1 (360. / (2. π)), σδMinFit1 (360. / (2. π))}, ", in degrees."]
        Print["The other hub, -Hmin , is located diametrically opposite from Hmin ."]
        Print["The uncertainty runs give the location of the avoidance hub Hmax  as  (α, δ)  = ",
         {αMaxFit (360. / (2. π)), δMaxFit (360. / (2. π))}, " ± ",
         {σαMaxFit (360. / (2. π)), σδMaxFit (360. / (2. π))}, ", in degrees."]
        Print["The other hub, -Hmax , is located diametrically opposite from Hmax ."]
        Print["The uncertainty runs determine the angle Θ between the two grey Great
           Circles in Fig. 5. to be  Θ = ", ΘminMAXFit, "° ± ", σΘminMAXFit, "°."]
```

To estimate the effects of experimental uncertainty, there were 5000 uncertainty runs.

Uncertainty runs have polarization directions $\psi = \psi n + \delta\psi$,
where $\delta\psi$ is chosen with a normal distribution of half-width $\sigma\psi$ about the best value ψn.

The uncertainty runs determine the smallest alignment angle to be $\bar{\eta}_{min}$ = 21.7226° ± 0.850515°.

The uncertainty runs determine the largest avoidance angle to be $\bar{\eta}_{max}$ = 66.0725° ± 0.921618°.

The uncertainty runs give the location for one of the alignment hub H_{min} as (α, δ) =
{189.688, -1.392} ± {2.20497, 2.43013}, in degrees.

The other hub, $-H_{min}$, is located diametrically opposite from H_{min}.

The uncertainty runs give the location of the avoidance hub H_{max} as (α, δ) =
{144.136, -24.9252} ± {19.6147, 13.6058}, in degrees.

The other hub, $-H_{max}$, is located diametrically opposite from H_{max}.

The uncertainty runs determine the angle Θ between
the two grey Great Circles in Fig. 5. to be Θ = 92.3208° ± 3.78289°.

In[∘]:=
```
mapOfσψHminHmax
Print["Figure 9: The ", Length[runData], " sets of hubs found for the uncertainty runs."]
Print["The alignment hubs Hmin and -Hmin are plotted as light blue dots, ", LightBlue, ". "]
Print["The avoidance hubs Hmax and -Hmax are plotted as pink dots, ", LightRed, "."]
Print["The most likely locations of the hubs are outlined in purple, ", Purple, "."]
```

Out[∘]=

The Hubs Found from the Uncertainty Runs



Figure 9: The 5000 sets of hubs found for the uncertainty runs.

The alignment hubs H_{min} and $-H_{min}$ are plotted as light blue dots, .

The avoidance hubs H_{max} and $-H_{max}$ are plotted as pink dots, .

The most likely locations of the hubs are outlined in purple, .

As a final image, we superimpose the map of the uncertainty run hubs $H_{\min}$, $-H_{\min}$, $H_{\max}$, and $-H_{\max}$ in Fig. 9 on the graph of the alignment angle function $\overline{\eta}(H)$, Fig. 5.

In[ ]:=

```
Show[{mapOfηBar, mapOfσψHminHmax}]
Print[
 "Figure 10: Overlay Fig. 9, Uncertainty Run Hubs, onto Fig. 5, Alignment Function η̄(H)
   using Best Values ψn.  Note that the light blue alignment hubs from the uncertainty
   runs closely follow the areas of convergence (blue) for the best values ψn. And
   the pink avoidance hubs follow the areas of extreme divergence (red). One sees
   that shifting the polarization directions slightly due to experimental uncertainty,
   shifts the locations of the hubs slightly. The shifted hubs favor areas, in blue
   and red, that are close to the extremes for the alignment function η̄(H) in Fig.5."]
```

Out[ ]=



Figure 10: Overlay Fig. 9, Uncertainty Run Hubs, onto Fig. 5, Alignment Function $\overline{\eta}(H)$ using Best Values ψn. Note that the light blue alignment hubs from the uncertainty runs closely follow the areas of convergence (blue) for the best values ψn. And the pink avoidance hubs follow the areas of extreme divergence (red). One sees that shifting the polarization directions slightly due to experimental uncertainty, shifts the locations of the hubs slightly. The shifted hubs favor areas, in blue and red, that are close to the extremes for the alignment function $\overline{\eta}(H)$ in Fig.5.

7. Concluding Remarks

The sample of QSOs studied in this notebook has percent polarizations above 0.6%. Polarized starlight in the region is planned to be studied in some future notebook. The data shows that polarized starlight in the region of these QSOs has much lower percent polarizations, about 0.1% or so. This suggests the Milky Way contribution is small. While comparing optical and radio polarization percentages could be innately faulty, it may be that the polarization for these radio QSOs originates with the QSO upon emission or has developed enroute or some mix of the two.

By the survey in Fig. 3, one sees that very significantly aligned regions are rare with QSOs. This is unlike polarized starlight sources in the Milky Way which has a large proportion of 5° regions well aligned, with $-\text{Log}_{10}(S)$ often over 9, when surveyed as in Fig. 3. While the percent polarization has a higher degree for QSOs compared with starlight, the significances of the alignments is generally much lower for QSOs compared with stars. It may be worthwhile to search the sky near the very significant regions, the color dots in Fig. 3, for objects that may have a polarizing effect on the radio waves from the QSOs.

If the alignment of the polarization directions of these 27 QSOs is due to some interaction with matter enroute, then the alignment hub H_{min} being near the sources on the sky surely entails a different physical situation than with other situations having hubs that are far from the sources. When the alignment is far from the sources, all sources are polarized in more or less the same direction. A hub close to the sample on the sky, as with the 27 QSOs here, could indicate a magnetic field in different directions for the different sources, yet organized in a way that produces very significant alignment. Whatever the successful explanations are, the explanation of polarization directions aligning with nearby hubs is expected to differ in some essential ways from explanations that fit alignment characterized by near equal position angles.

References

1. R. Shurtleff, the ready-to-run Mathematica version of this notebook is available at the following URL:
https://www.dropbox.com/s/10b3te0yib67xgr/20210419Clump1RA175Dec10ForViXra.nb?dl=0
2. Wolfram Research, Inc., Mathematica, Version 12.1, Champaign, IL (2020).
3 . Wikipedia contributors. “Aitoff projection.” Wikipedia, The Free Encyclopedia. Wikipedia, The Free Encyclopedia, 25 May. 2017. Web. (3 Jan. 2018).
4. R. Shurtleff, “Indirect polarization alignment with points on the sky, the Hub Test” , https://vixra.org/abs/2011.0026 (2020).
5. Hutsemékers, D., Braibant, L., Pelgrims, V., and Sluse, D., Alignment of quasar polarizations with large-scale structures, Astron. Astrophys., 572, A18, doi: 10.1051/0004-6361/201424631, arXiv:astro-ph:1409.6098 (2014).
6. Pelgrims, V. and Hutsemékers, D., Polarization alignments of quasars from the JVAS/CLASS 8.4-GHz surveys, MNRAS, 450, 4161-4173, doi: 10.1093/mnras/stv917, arXiv:astro-ph:1503.03482 (2015).
7. Jackson, N., Battye, R. A., Browne, I. W. A., Joshi, S., Muxlow, T. W. B., and Wilkinson, P. N., A survey of polarization in the JVAS/CLASS flat-spectrum radio source surveys - I. The data and catalogue production, MNRAS, 376, 371-377, doi: 10.1111/j.1365-2966.2007.11442.x , arXiv:astro-ph/0703273 (2007).

```
In[ ]:= Print["The date and time that this statement was evaluated: ", Now]

The date and time that this statement was evaluated: Sun 9 May 2021 14:51:46 GMT-4.
```